
第 1章

代数方程式 (fsolve)

1.1 概要

代数方程式の解 f(x)=0を数値的に求めることを考える．標準的な

二分法 (bisection method)とニュートン法 (Newton’s method)

の考え方と例を説明し，

収束性 (convergency)と安定性 (stability)

について議論する．さらに収束判定条件について言及する．

二分法のアイデアは単純．中間値の定理より連続な関数では，関数の符号が変わる二つの変数の間には根が必ず存

在する．したがって，この方法は収束性は決して高くはないが，確実．一方，Newton法は関数の微分を用いて収束

性を速めた方法である．しかし，不幸にして収束しない場合や微分に時間がかかる場合があり，初期値や使用対象に

は注意を要する．

1.2 Mapleでの解

Mapleでは代数方程式の解は，fsolveで求まる．

x2 − 4x+ 1 = 0

の解を考える．未知の問題では時として異常な振る舞いをする関数を相手にすることがあるので，先ずは関数の概形

を見ることを常に心がけるべき．

> restart;

> func:=x->x^2-4*x+1;

func := x 7→ x2 − 4x+ 1

> plot(func(z),z=-1..7);

もし，解析解が容易に求まるなら，その結果を使うほうがよい．Maple scriptの解析解を求める solveでは，

> solve(func(x)=0,x);

　　　　　　　　 　　　　　　　　

　　　　　　　　 　　　　　　　　

と即座に求めてくれる．数値解は以下の通り求められる．

> fsolve(func(x)=0,x);

　　　　　　　　 　　　　　　　　

　　　　　　　　 　　　　　　　　

1.3 二分法と Newton法の原理

1.3.1 二分法 (bisection)

二分法は領域の端 x1, x2 で関数値 f(x1), f(x2) を求め，中間の値を次々に計算して，解を囲い込んでいく方法で

ある．

> plot(func(z),z=0..0.8,gridlines=true);

x1 x2 f(x1) f(x2)

0.0 0.8 　　　　　 　　　　　

　　　　 　　　　 　　　　 　　　　

　　　　 　　　　 　　　　 　　　　

　　　　 　　　　 　　　　 　　　　

1.3.2 Newton法 (あるいは Newton-Raphson法)

Newton法は最初の点 x1 から接線をひき，それが x軸 (y=0)と交わった点を新たな点 x2 とする．さらにそこで

の接線を求めて...

という操作を繰り返しながら解を求める方法である．関数の微分を df(x)とすると，これらの間には

という関係が成り立つ．

> df:=unapply(diff(func(x),x),x);

> with(plots):with(plottools):

> x1:=1.0:x0:=0.0:

> p:=plot(func(z),z=0..1.1):

> p1:=plot(df(x1)*(z-x1)+func(x1),z=0..1.1,color=blue):

> p2:=[disk([x1,func(x1)],0.02), disk([x0,0],0.02)]:

> display(p,p1,p2,gridlines=true);

x1 f(x1) df(x1)

1.0　 　　　　 　　　　

　　　　　　　 　　　　　　　 　　　　　　　

　　　　　　　 　　　　　　　 　　　　　　　

1.4 二分法と Newton法のコード

1.4.1 二分法 (bisection)

> x1:=0: x2:=0.8:

f1:=func(x1): f2:=func(x2):

for i from 1 to 5 do

x:=(x1+x2)/2;

f:=func(x);

if f*f1>=0.0 then

x1:=x; f1:=f;

else

x2:=x; f2:=f;

end if;

printf("%20.15f, %20.15f\n",x,f);

end do:

0.400000000000000, -0.440000000000000

0.200000000000000, 0.240000000000000

0.300000000000000, -0.110000000000000

0.250000000000000, 0.062500000000000

0.275000000000000, -0.024375000000000

1.4.2 Newton法 (あるいは Newton-Raphson法)

> dfunc:=unapply(diff(func(z),z),z);

dfunc := z 7→ 2z − 4

> x:=1: f:=func(x):

printf("%15.10f, %+24.25f\n",x,f);

for i from 1 to 5 do

x:=x-f/dfunc(x);

f:=func(x);

printf("%15.10f, %+24.25f\n",x,f);

end do:

1.0000000000, -2.0000000000000000000000000

0.0000000000, +1.0000000000000000000000000

0.2500000000, +0.0625000000000000000000000

0.2678571429, +0.0003188775510000000000000

0.2679491900, +0.0000000084726737970000000

0.2679491924, +0.0000000000000000059821834

以下のように Digitsを変更すれば，Mapleでは浮動小数点演算の有効数字を変えることができる．

> Digits:=40;

40

1.5 収束性と安定性

実際のコードの出力からも分かる通り，解の収束の速さは 2つの手法で極端に違う．2分法では一回の操作で解の

区間が半分になる．このように繰り返しごとに誤差幅が前回の誤差幅の定数 (< 1)倍になる方法は 1次収束 (linear

convergence)するという．Newton法では関数・初期値が素直な場合 (f ′(x) <> 0)に，収束が誤差の 2乗に比例す

る 2次収束を示す．以下はその導出を示した．

> restart; ff:=subs(xi-x[f]=ei,series(f(xi),xi=x[f],4));

ff := f (xf ) +D (f) (xf ) ei +
1

2
D(2) (f) (xf ) ei

2 +
1

6
D(3) (f) (xf ) ei

3 +O
(
ei4
)

> dff:=subs({0=x[f],x=ei},series(diff(f(x),x),x,3));

dff := D (f) (xf ) +D(2) (f) (xf ) ei +
1

2
D(3) (f) (xf ) ei

2 +O
(
ei3
)

> ei1:=ei-ff/dff;

ei1 := ei −
f (xf ) +D (f) (xf ) ei +

1
2 D

(2) (f) (xf ) ei
2 + 1

6 D
(3) (f) (xf ) ei

3 +O
(
ei4
)

D (f) (xf ) +D(2) (f) (xf ) ei +
1
2 D

(3) (f) (xf ) ei
2 +O

(
ei3
)

> ei2:=simplify(convert(ei1,polynom));

ei2 :=
1

3

3D(2) (f) (xf ) ei
2 + 2D(3) (f) (xf ) ei

3 − 6 f (xf )

2D (f) (xf ) + 2D(2) (f) (xf ) ei +D(3) (f) (xf ) ei
2

> ei3:=series(ei2,ei,3);

ei3 := − f (xf )

D (f) (xf )
+

f (xf )
(
D(2)

)
(f) (xf ) ei

(D (f) (xf ))
2 +

1

6

3
(
D(2)

)
(f) (xf ) + 3

f(xf )(D(3))(f)(xf )

D(f)(xf )
− 6

f(xf )((D(2))(f)(xf ))
2

(D(f)(xf ))
2

(D (f) (xf ))
ei2 +O

(
ei3
)

(1.1)

> subs(f(x[f])=0,ei3);

1

2

D(2) (f) (xf ) ei
2

D (f) (xf )
+O

(
ei3
)

注意すべきは，この収束性には一回の計算時間の差は入っていないことである．Newton法で解析的に微分が求まら

ない場合，数値的に求めるという手法がとられるが，これにかかる計算時間はばかにできない．二分法を改良した割

線法 (secant method)がより速い場合がある (NumRecipe9章参照)．

二分法では，収束は遅いが，正負の関数値の間に連続関数では必ず解が存在するという意味で解が保証されている．

しかし，Newton法では，収束は速いが，必ずしも素直に解に収束するとは限らない．解を確実に囲い込む，あるい

は解に近い値を初期値に選ぶ手法が種々考案されている．解が安定であるかどうかは，問題，解法，初期値に大きく

依存する．収束性と安定性のコントロールが数値計算のツボとなる．

1.6 収束判定条件

どこまで値が解に近づけば計算を打ち切るかを決める条件を収束判定条件と呼ぶ．以下のような条件がある．

ε(イプシロン，epsilon)法

δ(デルタ，delta)法

占部法 数値計算の際の丸め誤差までも含めて判定する条件で，

|f(xi+1)| > |f(xi)|
とする．

■ϵ, δ を説明するための図 　

> with(plots):with(plottools):

> f2:=x->0.4*(x^2-4*x+1):x1:=0.25:x0:=0.4:

p1:=plot([f2(z),f2(x1)],z=0.2..0.5):

p2:=[disk([x1,f2(x1)],0.005),disk([x0,f2(x0)],0.005)]:

l1:=line([x0,f2(x0)],[x0,f2(x1)]):

t1 := textplot([0.45,0.0,‘epsilon‘],align=above):

t2 := textplot([0.325,0.05,‘delta‘],align=below):

> display(p,p1,p2,l1,t1,t2);

1.7 2変数関数の場合

2変数の関数では，解を求める一般的な手法は無い．この様子は実際に 2変数の関数で構成される面の様子をみれ

ば納得されよう．

> restart;

> f:=(x,y)->4*x+2*y-6*x*y; g:=(x,y)->10*x-2*y+1;

f := (x, y) 7→ 4x+ 2 y − 6xy

g := (x, y) 7→ 10x− 2 y + 1

> p1:=plot3d({f(x,y)},x=-2..2,y=-2..2,color=red):

p2:=plot3d({g(x,y)},x=-2..2,y=-2..2,color=blue):

p3:=plot3d({0},x=-2..2,y=-2..2,color=gray):

with(plots):

display([p1,p2,p3],axes=boxed,orientation=[-150,70]);

解のある程度近くからは，Newton法で効率良く求められる．

> fsolve({f(x,y)=0,g(x,y)=0},{x,y});

{x = −0.07540291160, y = 0.1229854420}

1.8 例題:二分法と Newton法の収束性

代数方程式に関する次の課題に答えよ．(2004年度期末試験)

1. exp(−x) = x2 を二分法およびニュートン法で解け.

2. n回目の値 xn と小数点以下 10桁まで求めた値 xf = 0.7034674225との差∆xn の絶対値 (abs)の logを nの

関数としてプロットし，その収束性を比較せよ．また，その傾きの違いを両解法の原理から説明せよ.

解答例

計算精度を 40桁にしておく．funcで関数を定義．

> restart; Digits:=40: func:=unapply(exp(-x)-x^2,x);

func := x 7→ e−x − x2

先ずは，関数を plotして概形を確認．

> plot(func(x),x=-5..5,y=-10..10);

Mapleの組み込みコマンドで正解を確認しておく．

> x0:=fsolve(func(x)=0,x);

0.7034674224983916520498186018599021303429

テキストからプログラムをコピーして走らせてみる．環境によっては，printf分の中の”\”が文字化けしているので

修正．

> x1:=0: x2:=0.8: res1:=[]:

f1:=func(x1): f2:=func(x2):

for i from 1 to 20 do

x:=(x1+x2)/2;

f:=func(x);

if f*f1>=0.0 then

x1:=x; f1:=f;

else

x2:=x; f2:=f;

end if;

printf("%20.15f, %20.15f\n",x,f);

res1:=[op(res1),[i,abs(x-x0)]]:

end do:

0.400000000000000, 0.510320046035639

0.600000000000000, 0.188811636094026

0.700000000000000, 0.006585303791410

0.750000000000000, -0.090133447258985

0.725000000000000, -0.041300431044638

中略

0.703468322753906, -0.000001712107681

0.703467559814453, -0.000000261147873

プロットのためにリストを res1で作成している．

同様に Newton法での結果を res2に入れる．

> dfunc:=unapply(diff(func(z),z),z);

dfunc := z 7→ −e−z − 2 z

> x:=1.0: f:=func(x):

printf("%15.10f, %+24.25f\n",x,f);

res2:=[[1,abs(x-x0)]]:

for i from 2 to 5 do

x:=x-f/dfunc(x);

f:=func(x);

printf("%15.10f, %+24.25f\n",x,f);

res2:=[op(res2),[i,abs(x-x0)]]:

end do:

1.0000000000, -0.6321205588285576784044762

0.7330436052, -0.0569084480040254074684576

0.7038077863, -0.0006473915387465014761973

0.7034674683, -0.0000000871660305624231097

0.7034674225, -0.0000000000000015809178420

res1, res2を片対数プロットして同時に表示．

> l1:=logplot(res1);

> l2:=logplot(res2);

> display(l1,l2);

2分法で求めた解は，Newton法で求めた解よりもゆっくりと精密解へ収束している．これは，二分法が原理的に計

算回数について一次収束なのに対して，Newton 法は 2 次収束であるためである．解の差 (δ) だけでなく，関数値

f(x), ϵをとっても同様の振る舞いを示す．

1.9 課題

1. Newton法の f(x), df(x)の関係を示す式を導け．

2. 次の関数 f(x) = exp(−x)− 2 exp(−2x) の解を二分法，Newton法で求めよ．

3. 代数方程式に関する次の課題に答えよ．(2004年度期末試験)

（a）exp(−x) = x2 を二分法およびニュートン法で解け.

（b）n回目の値 xn と小数点以下 10桁まで求めた値 xf = 0.7034674225との差 ∆xn の絶対値 (abs)の logを

nの関数としてプロットし，その収束性を比較せよ．また，その傾きの違いを両解法の原理から説明せよ.

4. 次の方程式 f(x) = x4 − x− 0.12 の正数解を二分法で求めよ．(2008年度期末試験)

5. 収束条件がうまく機能しない例を示せ．

6. 割線法は，微分がうまく求まらないような場合に効率がよい，二分法を改良した方法である．二分法では新た

な点を元の 2点の中点に取っていた．そのかわりに下図に示すごとく，新たな点を元の 2点を直線で内挿した

点に取る．二分法のコードを少し換えて，割線法のコードを書け．また，収束の様子を二分法，Newton法と

比べよ．

> func:=x->x^2-4*x+1: x1:=0: x2:=2: f1:=func(x1): f2:=func(x2):

> plot({(z-x1)*(f1-f2)/(x1-x2)+f1,func(z)},z=0..2);

7. 次の方程式 f(x) = cos(x)− x2 の正数解を二分法で求めよ．割線法でも求め，収束性を比べよ．(2009年度期

末試験)

8. 次の方程式 f(x) = x3 − 3x + 3の解をニュートン法で求めよ．初期値をそれぞれ x = −3, x = 2とした時を

比べ，その差について論ぜよ．(2010年度期末試験)

第 2章

誤差 (Error)

2.1 打ち切り誤差と丸め誤差 (Truncation and round off errors)

数値計算のねらいは，できるだけ精確・高速に解を得ることである．誤差 (精度) と収束性 (安定性,速度) が数値

計算のキモとなる．前回に説明した収束判定条件による誤差は打ち切り誤差 (truncation error) と呼ばれる．ここで

は，誤差のもう一つの代表例である，計算機に特有の丸め誤差 (roundoff error) について見ておこう．

2.1.1 整数型と実数型の内部表現

計算機は一般に無限精度の計算をおこなっているわけではない．CPU で足し算をおこなう以上，一般的な計算に

おいては CPU が扱う一番効率のいい数の大きさが存在する．これが,32bit の CPU では 1 ワード,4byte(4x8bits)

である．1ワードで表現できる最大整数は,符号に 1bit 必要なので,2^(31)-1 となる．実数は以下のような仮数部と

指数を取る浮動小数点数で表わされる．

表 2.1 浮動小数点数の内部表現 (IEEE754)．

s× f ×Be−E

s sign bit(符号ビット:正負の区別を表す)

e biased exponent(指数部)

f fraction portion of the number(仮数部)

B base(基底) で通常は 2　

E bias(下駄)と呼ばれる

real(単精度) s=1, e=8, f=23 E=127

double precision(倍精度) s=1, e=11, f=52 E=1023

Eは指数が負となる小数点以下の数を表現するためのもの．演算結果は実際の値から浮動小数点数に変換するため

の操作「丸め (round-off)」が常に行われる．それに伴って現れる誤差を丸め誤差と呼ぶ．

2.2 有効桁数 (Significant digits)

1 ワードの整数の最大値とその 2進数表示．

> restart;

> 2^(4*8-1)-1;#res: 2147483647

この整数を 2進数で表示するように変換するには，convert(n,binary)を用いて，

> convert(2^(4*8-1)-1,binary); #res: 1111111111111111111111111111111

となり，31 個の 1 が並んでいることが分かる．1 ワードの整数の最大桁は，n の桁数を戻すコマンド length(n) を

使って，

> length(2^(4*8-1)-1);　#res: 10

となり，たかだか 10桁程度であることが分かる．一方，64bit の場合の整数の最大桁．

> length(2^(8*8-1)-1);　#res: 19

である．

Maple では多倍長計算するので，通常のプログラミング言語で起こる intの最大数あたりでの奇妙な振る舞いは示

さない．

> 2147483647+100; #res: 2147483747

単精度の浮動小数点数は,仮数部 2進数 23bit,2倍長実数で 52bitである．この有効桁数は以下の通り．

> length(2^(23)); #res: 7

> length(2^(52)); #res: 16

2.3 浮動小数点演算による過ち (FloatingPointArithmetic)

「丸め」にともなって誤差が生じる. Cや Fortran等の通常のプログラミング言語では「丸める」仕様なのでプログ

ラマーが気をつけなければならない．

プログラムリスト : 実数のケタ落ち

#include <stdio.h>

int main(void){

float a,b,c;

double x,y,z;

a=1.23456789;

printf(" a= %17.10f\n",a);

b=100.0;

c=a+b;

printf("%20.10f %20.10f %20.10f\n",a,b,c);

x=(float)1.23456789;

y=(double)100;

z=x+y;

printf("%20.12e %20.12e %20.12e\n",x,y,z);

x=(double)1.23456789;

y=(double)100;

z=x+y;

printf("%20.12e %20.12e %20.12e\n",x,y,z);

return 0;

}

分かっているつもりでも，よくやる間違い．

プログラムリスト : 丸め誤差

#include <stdio.h>

int main(void){

float x=77777,y=7,y1,z,z1;

y1=1/y;

z=x/y;

z1=x*y1;

printf("%10.2f %10.2f\n",z,z1);

if (z!=z1){

printf("z is not equal to z1.\n");

}

printf("Surprising?? \n\n\n\n\n%10.5f %10.5f\n",z,z1);

return 0;

}

これを避けるには，EPSILONという小さな数字を定義しておいて，値の差の絶対値を求める fabsを使って

とすべき．このときは数学関数である fabsを使っているので，

> gcc -lm test.c

とmath libraryを明示的に呼ぶのを忘れないように．

2.4 機械精度 (Machine epsilon)

上の例では，浮動小数点数で計算した場合に小さい数の差を区別することができなくなるということを示している．

これは，CPUに固有の精度で，機械精度 (Machine epsilon)と呼ばれる．つまり，小さい数を足したときにその計算

機がその差を認識できなくなる限界ということで，以下のようにして求めることができる．

> Digits:=7;

> e:=evalf(1.0);

> w:=evalf(1.0+e);

> while (w>1.0) do

printf("%-15.10e %-15.10e %-15.10e\n",e,w,evalf(w-1.0));

e:=evalf(e/2.0);

w:=evalf(1.0+e);

end do:

7

1.0

2.0

1.0000000000e+00 2.0000000000e+00 1.0000000000e+00

5.0000000000e-01 1.5000000000e+00 5.0000000000e-01

2.5000000000e-01 1.2500000000e+00 2.5000000000e-01

1.2500000000e-01 1.1250000000e+00 1.2500000000e-01

6.2500000000e-02 1.0625000000e+00 6.2500000000e-02

3.1250000000e-02 1.0312500000e+00 3.1250000000e-02

1.5625000000e-02 1.0156250000e+00 1.5625000000e-02

7.8125000000e-03 1.0078120000e+00 7.8120000000e-03

3.9062500000e-03 1.0039060000e+00 3.9060000000e-03

1.9531250000e-03 1.0019530000e+00 1.9530000000e-03

9.7656250000e-04 1.0009770000e+00 9.7700000000e-04

4.8828120000e-04 1.0004880000e+00 4.8800000000e-04

2.4414060000e-04 1.0002440000e+00 2.4400000000e-04

1.2207030000e-04 1.0001220000e+00 1.2200000000e-04

6.1035150000e-05 1.0000610000e+00 6.1000000000e-05

3.0517580000e-05 1.0000310000e+00 3.1000000000e-05

1.5258790000e-05 1.0000150000e+00 1.5000000000e-05

7.6293950000e-06 1.0000080000e+00 8.0000000000e-06

3.8146980000e-06 1.0000040000e+00 4.0000000000e-06

1.9073490000e-06 1.0000020000e+00 2.0000000000e-06

9.5367450000e-07 1.0000010000e+00 1.0000000000e-06

2.5 桁落ち，情報落ち，積み残し (Cancellation)

■桁落ち (Cancellation) 　
0.723657

− 0.723649

■情報落ち (Loss of Information) 　

72365.7
+ 1.23659

■積み残し 　
72365.7

+ 0.001

2.6 課題

1. 次の項目について答えよ．(2004, 05, 06年度期末試験)

（a）数値計算の精度を制約するデータ形式とその特徴は何か．

（b）丸め誤差とは何か．

（c）打ち切り誤差とは何か．

（d）安定性とは何か．

2. 10 進数 4 桁の有効桁数をもった計算機になったつもりで，a=0.81543, b=35.724, c=35.596として，以下の

計算をおこなえ．

（a）b-c

（b）a/(b-c)

3. 10 進数 3 桁の有効桁数をもった計算機になったつもりで，以下の条件で預金を求める計算をおこなえ．

（a）元本を 10000円とする

（b）利息 0.3%とする

（c）複利計算で 10年でいくらになるか．

4. 係数を a = 1, b = 10000000, c = 1としたときに, 通常の解の公式を使った解と, 解と係数の関係 (下記の記

述を参照）を使った解とを出力するプログラムをMapleで作成し, 2種類計算方法の違いで解の有効数字がど

う違うかを比べよ．

2 次方程式 ax2 + bx+ c = 0の係数 a, b, cが特殊な値をもつ場合，通常の解の公式

x =
−b±

√
b2 − 4ac

2a

にしたがって計算するとケタ落ちによる間違った答えを出す．その特殊な値とは√
b2 − 4ac ≈ |b|

となる場合である．

ケタ落ちを防ぐには, b > 0の場合は,

x1 =
−b−

√
b2 − 4ac

2a

として，ケタ落ちを起こさずに求め, この解を使って, 解と係数の関係より

x2 =
c

a x1

で求める．b < 0 の場合は，解の公式の足し算の方を使って同様に求める．

第 3章

線形代数–写像 (LAFundamentals)

3.1 行列と連立方程式

大学の理系で必修なのは微積分と線形代数です．線形代数というと逆行列と固有値の計算がすぐに思い浮かぶで

しょう．計算がややこしくてそれだけでいやになります．でも，行列の計算法は一連の手順で記述できるので，Maple

では微積分とおなじように一個のコマンドで片が付きます．それが 3x3以上でも同じです．問題はその意味です．こ

こでは，線形代数の計算がMapleを使えばどれほど簡単にできるかを示すと共に，線形代数の基本となる概念につい

てスクリプトと描画を使って，直観的に理解することを目的とします．

先ずは連立方程式から入っていきます．中学の時に

4x = 2

というのを解きますよね．一般的には
ax = b
x = b/a

と書けるというのは皆さんご存知のはず．これと同じようにして連立方程式を書こうというのが逆行列の基本．つ

まり
2x+ 5y = 7
4x+ y = 5

という連立方程式は，係数から作られる 2x2行列を係数行列 A，左辺の値で作るベクトルを bとして，

Ax = b
x = b/A = A−1b

としたいわけです．

実際にMapleでやってみましょう．行列は英語でMatrixです．

> restart: A:=Matrix([[2,5],[4,1]]);

A :=

[
2 5
4 1

]
こうして行列を作ります．

> b:=Vector([7,5]); #(2)ベクトルは英語で Vectorです．これで縦ベクトルができます．

b :=

[
7
5

]
線形代数は linear algebra と言います．with で LinearAlgebra というライブラリーパッケージを読み込んでおき

ます．

> with(LinearAlgebra):

逆行列は matrix inverseと言います．

> x0:=MatrixInverse(A).b;

行列 AのMatrixInverseを求めて，ベクトル bに掛けています．

x0 :=

[
1
1

]
と簡単に求めることができます．

3.2 掃き出し

係数行列 Aとベクトル bを足して作られる行列は拡大係数行列と呼ばれます．Mapleでは，これは

> <A|b>; [
2 5 7
4 1 5

]
として作られます．ここから行列の掃き出し操作をおこなうには，LUDecompositionというコマンドを使います．

> P,L,U:=LUDecomposition(<A|b>);

P, L, U :=

[
1 0
0 1

]
,

[
1 0
2 1

]
,

[
2 5 7
0 −9 −9

]
これは，下三角行列 (Lower Triangle Matrix)と上三角行列 (Upper Triangle Matrix)に分解 (decompose)するコ

マンドです．P 行列は置換 (permutation)行列を意味します．LUDecompositionだけでは，前進消去が終わっただ

けの状態です．そこで，後退代入までおこなうには，optionに output=’R’をつけます．そうすると出力は，

> LUDecomposition(<A|b>,output=’R’); [
1 0 1
0 1 1

]
で，bベクトルの部分が解になっています．

3.3 写像

次に，これを 2 次元上のグラフで見てみましょう．先ず描画に必要なライブラリーパッケージ (plots および

plottools)を withで読み込んでおきます．

> with(plots):with(plottools):

ベクトルは，位置座標を意味するように listへ変換 (convert)しておきます．

> p0:=convert(x0,list); p1:=convert(b,list);

位置 p0に円 (disk)を半径 0.2,赤色で描きます．同じように位置 p1に半径 0.2，青色で diskを描きます．もう一つ，

p0から p1に向かう矢印 (arrow) を適当な大きさで描きます．後ろの数字をいじると線の幅や矢印の大きさが変わり

ます．

> point1:=[disk(p0,0.2,color=red), disk(p1,0.2,color=blue)]:

> line1:=arrow(p0,p1,.05,.3,.1 ):

これらをまとめて表示 (display)します．このとき，表示範囲を-8..8,-8..8とします．

> display(point1,line1,view=[-2..8,-2..8],gridlines=true);

逆行列は

> MatrixInverse(A); [
−1/18 5/18
2/9 −1/9

]
で求まります．先ほどの矢印を逆に青から赤へたどる変換になっています．これが，連立方程式を解く様子をグラフ

で示しています．つまり，行列 Aで示される変換によって求まる青点で示したベクトル b(7,5)を指す元の赤点を捜

すというものです．答えは (1,1)となります．

では，元の赤点をもう少しいろいろ取って，行列 Aでどのような点へ写されるかを見てみましょう．

> N:=30:point2:=[]:line2:=[]:

for k from 0 to N-1 do

x0:=Vector([sin(2*Pi*k/N),cos(2*Pi*k/N)]);

x1:=A.x0;

p0:=convert(x0,list);

p1:=convert(x1,list);

point2:=[op(point2),disk(p0,0.05,color=red)];

point2:=[op(point2),disk(p1,0.05,color=blue)];

line2:=[op(line2),line(p0,p1)];

end do:

N:=30で分割した円周上の点を x0で求めて，point2にその円とそれのＡ.x0を，line2にはその 2点を結ぶ line(線)を

足しています．使っているコマンドは，先ほどの描画とほぼ同じです．ただし，Mapleスクリプトに特有の idiom(熟

語)を使っています．この基本形をとり出すと，

> list1:=[];

for k from 0 to 2 do

list1:=[op(list1),k];

end do;

list1;

[]

[0]

[0, 1]

[0, 1, 2]

[0, 1, 2]

となります．for-loopで kを 0から 4まで回し，list1に次々と値を追加していくというテクです．

できあがりの次の図を見てください．

> d:=6: display(point2,line2,view=[-d..d,-d..d]);

何やっているか分かります? 中心の赤点で示される円が，青点で示される楕円へ写されていることが分かるでしょ

うか．

線形代数の講義で，写像を示すときによく使われるポンチ絵を現実の空間で示すとこのようになります．ポンチ絵

では，赤で示した V 空間が青で示したW 空間へ行列 Aによって写像され，それぞれの要素 v が w へ移されると意

図しています．

V W

3.4 固有ベクトルの幾何学的意味

では，ここでクイズです．固有ベクトルは上のグラフの何処に対応するか?　ヒントは，

行列 Aの固有値，固有ベクトルを λ, x0 とすると，

Ax0 = λx0

が成立する

です．固有値と固有ベクトルはMapleでは以下のコマンドで求まります．

> lambda,P:=Eigenvectors(A);

λ, P :=

[
−3
6

]
,

[
−1 5/4
1 1

]
ここではMapleコマンドの Eigenvectorsで戻り値を λ(lambdaと書きます)，P に代入しています．この後ろ側に

ある行列 P の 1列目で構成されるベクトルが固有値-3に対応する固有ベクトル，2列目のベクトルが固有値 6に対応

する固有ベクトルです．

3.4.1 解答

固有値 λ，固有ベクトル x0 の関係式
Ax0 = λx0

を言葉で言い直すと，

固有ベクトル x0 は変換行列 Aによって，自分の固有値倍のベクトル λx0 に写されるベクトル

となります．つまり変換の図で言うと，

変換しても方向が変わらない赤点（の方向）

となります．これは図で書くと

> vv1:=Column(P,1): vv2:=Column(P,2):

a1:=vv1[2]/vv1[1]: a2:=vv2[2]/vv2[1]:

pp1:=plot({a2*x,a1*x},x=-d..d):

Columnによって行列の第 i列目をとりだし，その比によって直線の傾きを求めています．そうして引いた 2本の直

線を pp1としてため込んで，先ほど描いた変換の図に加えて表示 (display)させます．

> display(point2,line2,pp1,view=[-d..d,-d..d]);

pp1を入れて描いた直線が引かれた方向ではたしかに変換によっても方向が変わらなさそうに見えるでしょう．

おまけですが，行列の対角化は次のようにしてできます．

> MatrixInverse(P).A.P; [
−3 0
0 6

]

3.5 行列式の幾何学的意味

行列 Aの行列式 (|A|あるいは detAと表記)は Determinantで求まります．

> Determinant(A);

−18

では次のクイズ．先ほど求めた，行列 Aの行列式は，どこに対応するでしょう? 以下の (1,0),(0,1)の点を変換した

点に原点からベクトルを結んでその意味を説明してください．さらに，そのマイナスの意味は？．

> point3:=[]:line3:=[]: XX:=Matrix([[1,0],[0,1]]):

for i from 1 to 2 do

x0:=Column(XX,i); x1:=A.x0;

p0:=convert(x0,list):

p1:=convert(x1,list):

point3:=[op(point3),disk(p0,0.2,color=red),disk(p1,0.2,color=blue)]:

line3:=[op(line3),arrow([0,0],p0,.05,.3,.1 ),arrow([0,0],p1,.05,.3,.1 )]:

end do:

display(point3,line3,view=[-1..8,-1..8],gridlines=true);

3.6 行列式が 0の写像

では，行列式が０になるというのはどういう状態でしょう? 次のような行列を考えてみましょう．

> A:=Matrix([[2,1],[4,2]]);

この行列式は

> Determinant(A);

0

です．この変換行列で，上と同じように写像の様子を表示させてみましょう．

> N:=30:point2:=[]:line2:=[]:

for k from 0 to N-1 do

x0:=Vector([sin(2*Pi*k/N),cos(2*Pi*k/N)]); x1:=A.x0; p0:=convert(x0,list);

p1:=convert(x1,list);

point2:=[op(point2),disk(p0,0.05,color=red),disk(p1,0.05,color=blue)];

line2:=[op(line2),line(p0,p1)];

end:

> d:=6: display(point2,line2,view=[-d..d,-d..d]);

わかります？

今回の移動先の青点は直線となっています．つまり，determinantが 0ということは，変換すると面積がつぶれる

という事を意味しています．平面がひとつ次元を落として線になるということです．

次に，この行列の表わす写像によって原点 (0,0)に写される元の座標を求めてみます．連立方程式に戻してみると

> A.Vector([x,y])=Vector([0,0]); [
2x+ y
4x+ 2 y

]
=

[
0
0

]
となります．とよく見ると，1行目も 2行目もおなじ式になっています．2次元正方行列で，行列式が 0の時には必

ずこういう形になり，直線の式となります．これを表示すると

> plot([-2*x,-2*x+1,-2*x-1],x=-4..4,y=-4..4);

> plot([2*x],x=-4..4,y=-4..4,color=blue);

左図の赤線となります．この直線上の全ての点が [0,0]へ写されることを確認してください．また，緑の線上の点は

全て [1,2]へ写されることが確認できます．

> A.Vector([-1,2]); [
0
0

]

こうしてすべて調べていけば，左の平面上のすべて点は右の青の直線上へ写されることが分かります．今まで見て

きた円と楕円とはまったく違った写像が，行列式が 0の行列では起こっていることが分かると思います．右の青線を

行列 Aによる像 (Image, ImAと表記)，左の赤線，つまり写像によって [0,0]へ写される集合を核 (Kernel, KerAと

表記)と呼びます．

これをポンチ絵で描くと，次の通りです．

像 (Image) 核 (Kernel)

V W

Im A

V W

Ker A

O

3.7 全単射

行列 Aによる写像を f として，赤点に限らず元の点の集合を V , 移った先の点の集合をW とすると，

f : V → W

と表記されます．v, w を V,W の要素としたとき，異なる v が異なる w に写されることを単射，全ての w に対応す

る v がある写像を全射と言います．全単射，つまり全射でかつ単射，だと要素は一対一に対応します．先ほどの Aは

全射でもなく，単射でもない例です．

行列式が 0の場合の写像は単射ではありません．このとき，逆写像が作れそうにありません．これを連立方程式に

戻して考えましょう．もともと，
v = A−1w

の解 v は点 w が写像 Aによってどこから写されてきたかという意味を持ちます．逆写像が作れない場合は，連立方

程式の解はパラメータをひとつ持った複数の解 (直線)となります．これが係数行列の行列式が 0の場合に，連立方程

式の解が不定となる，あるいは像がつぶれるという関係です．

行列の次元が高い場合には，いろいろなつぶれかたをします．行列の階数と次元は

> Rank(A);

Dimension(A);

1

2, 2

で求まります．

Aをm行 n列の行列とするとき，

Rank(A) = Dimension (Im A)

Dimension (Ker A) = n - Rank(A)

が成立し，これを次元定理といいます．全射と単射の関係は，下の表のような一変数の方程式での解の性質の拡張

と捉えることができます．

表 3.1 代数方程式 ax = bの解の存在性．

一意 a <> 0 x = b/a

不定 a = 0, b = 0 解は無数

不能 a = 0, b <> 0 解は存在しない

表 3.2 連立方程式 Ax = bの解の存在性．

m× n行列 A 全射でない (Im A < m), 値域

上にあるときのみ解が存在

全射 (ImA = m), 解は必ず存

在

単射でない (Ker A <> 0), 解

は複数 V W

?

V W

単射 (Ker A = 0), 解はひとつ

V W

?

V W

3.8 課題

1. 下の図は A :=

[
3 2/3

2/3 2

]
を用いて変換される像を表わしている．この行列の固有値，行列式が何処に対

応するか説明せよ．また，固有ベクトルの方向を記せ．(2007年度期末試験)

4.1 行列計算の概要

数値計算の中心課題の一つである，行列に関する演算について見ていく．多次元，大規模な行列に対する効率のよ

い計算法が多数開発されており，多くの既存のライブラリが用意されている．本章ではそれらの中心をなす，逆行列

(matrix inverse)と固有値 (Eigen values)に関して具体的な計算方法を示す．現実的な問題には既存のライブラリを

使うのが上策であるが，それでも基礎となる原理の理解や，ちょっとした計算，ライブラリの結果の検証に使える

ルーチンを示す．

逆行列は連立一次方程式を解くことと等価である．ルーチン的なやり方にガウスの消去法がある．これは上三角行

列になれば代入を適宜おこなうことで解が容易に求まることを利用する．さらに，初期値から始めて次々に解に近づ

けていく反復法がある．この代表例である Jacobi(ヤコビ)法と，収束性を高めた Gauss-Seidel(ガウス- ザイデル)法

を紹介する．

上記の手法をより高速にした修正コレスキー分解と共役傾斜 (共役勾配) 法があるが，少し複雑になるので割愛す

る．必要ならば NumRecipeを読め．

4.2 ガウス消去法による連立一次方程式の解

逆行列は連立一次方程式を解くことと等価である．すなわち，Aを行列，xを未知数ベクトル，bを数値ベクトル

とすると，
Ax = b

A−1Ax = A−1b
x = A−1b

である．未知数の少ない連立一次方程式では，適当に組み合わせて未知数を消していけばいいが，未知数が多くなっ

てしまうと破綻する．未知数の多い多元連立一次方程式で，ルーチン的に解を求めていく方法がガウス消去法で，前

進消去と後退代入という 2つの操作からなる．

後退代入 (Backward substitution)による解の求め方を先ず見よう．たとえば，

x+ y − 2z = −4
−3y + 3z = 9

−z = −2

では，下から順番に z → y → xと適宜代入することによって，簡単に解を求めることが出来る．係数で作る行列で

このような形をした上三角行列にする操作を前進消去あるいはガウスの消去法 (Gaussian elimination)という．下三

角行列 L(lower triangular matrix)と上三角行列 U(upper triangular matrix)の積に分解する操作

A = L.U

を LU分解 (LU decomposition)という．例えば先に示した上三角行列を係数とする連立方程式は，

x+ y − 2z = −4
x− 2y + z = 5
2x− 2y − z = 2

を変形することで得られる．この変形を示せ．

4.3 Mapleによる LU分解

係数行列 (coefficient matrix)と定数項 (b)との関係は以下の通りである．

> restart;

A:=Matrix([[1,1,-2],[1,-2,1],[2,-2,-1]]):

X:=Vector([x,y,z]):

#X:=Vector([1,-1,2]):

b:=Vector([-4,5,2]):

A.X=b;  x+ y − 2 z
x− 2 y + z
2x− 2 y − z

 =

 −4
5
2


単に逆行列を求める際は

> with(LinearAlgebra):

MatrixInverse(A);  4/3 5/3 −1
1 1 −1

2/3 4/3 −1


である．Mapleでは行列を三角行列に分解するために，LUDecompositionコマンドが用意されている．

> P,L,U:=LUDecomposition(<A|b>);

P, L, U :=

 1 0 0
0 1 0
0 0 1

 ,

 1 0 0
1 1 0
2 4/3 1

 ,

 1 1 −2 −4
0 −3 3 9
0 0 −1 −2


係数と定数項から作られる行列を拡大係数行列 (augmented matrix) といい，Maple では，< A|b > で作られる．

後退代入までおこなって連立方程式の解を求めるには，以下の通り optionに output=’R’をつける．

> LUDecomposition(<A|b>,output=’R’);  1 0 0 1
0 1 0 −1
0 0 1 2



4.4 LU分解のコード

LU分解すれば線形方程式の解が容易に求まることは理解できると思う．具体的に Aを LU分解する行列 (消去行

列と称す)T1,T2の係数は次のようにして求められる．

> A0:=Matrix([[1,1,-2],[1,-2,1],[2,-2,-1]]):

b0:=Vector([-4,5,2]):

A:=Matrix(A0): B:=Vector(b0): n:=3:

L:=Matrix(array(1..n,1..n,identity)):

for i from 1 to n do #i行目

T[i]:=Matrix(array(1..n,1..n,identity)):

#i番目の消去行列を作る

for j from i+1 to n do

am:=A[j,i]/A[i,i]; #i行の要素を使って，i+1行目の先頭を消す係数を求める

T[i][j,i]:=-am; #i番目の消去行列に要素を入れる

L[j,i]:=am; #LTMの要素

for k from 1 to n do

A[j,k]:=A[j,k]-am*A[i,k]; #もとの行列を UTMにしていく

end do;

B[j]:=B[j]-B[i]*am; #数値ベクトルも操作

end do;

end do:

上のコードによって得られた消去行列．

> T[1]; T[2];

 1 0 0
−1 1 0
−2 0 1


 1 0 0

0 1 0
0 −4/3 1


これを実際に元の行列 A0に作用させると，UTMが求められる．

> U:=T[2].T[1].A0;

U :=

 1 1 −2
0 −3 3
0 0 −1


求められた LTM, UTMを掛けると

> L.U;  1 1 −2
1 −2 1
2 −2 −1


元の行列を得られる．L,Aに求めたい行列が入っていることを確認．

> L;A;

1 1 0
2 4/3 1


 1 1 −2

0 −3 3
0 0 −1


数値ベクトルも期待通り変換されている．

> B;  −4
9
−2



4.5 ピボット操作

ガウス消去法で困るのは，割ろうとした対角要素が 0の場合である．しかし，この場合にも，方程式の順序を，行

列の行と右辺の値をペアにして入れ替えれば解決する．この割るほうの要素をピボット要素あるいはピボット (pivot,

バスケの軸足を動かさずにくるくる回すやつ)と呼ぶ．この操作は，変数の並びを変えたわけではなく，単に方程式

の順番を変更する操作に相当する．

さらに対角要素の数値が厳密に 0でなくとも，極端に 0に近づいた場合にも，その数で割った数値が大きくなり他

の数との差を取ると以前に示した情報落ちの可能性が出てくる．この現象を防ぐためには，絶対値が最大のピボット

を選んで行の入れ替えを毎回おこなうといい結果が得られることが知られている．

Maple の LUDecomposition コマンドをこのような行列に適用すると，置換行列 (permutation matrix)P が単位

行列ではなく，ピボット操作に対応した行列となる．P.A=L.Uとなることに注意．

4.6 反復法による連立方程式の解

以下のような連立方程式を 
5x+ y + z + u
x+ 3 y + z + u
x− 2 y − 9 z + u
x+ 3 y − 2 z + 5u

 =


−6
2
−7
3


形式的に解くと

x =
−6− (y + z + u)

5

となる．他の未知数も，

y=

z=

u=

となる．適当に初期値 (x0, y0, z0, u0)をとり，下側の方程式に代入すると，得られた出力 (x1, y1, z1, u1)はより正解

に近い値となる．これを繰り返すことによって正解が得られる．これをヤコビ (Jacobi)法と呼び，係数行列の対角要

素が非対角要素にくらべて大きいときに適用できる．多くの現実の問題ではこの状況が成り立っている．

Gauss-Seidel法は Jacobi法の高速版である．n番目の解の組が得られた後に一度に次の解の組に入れ替えるので

はなく，得られた解を順次改良した解として使っていく．これにより，収束が早まる．以下にはヤコビ法のコードを

示した．x1[i]の配列を変数に換えるだけで，Gauss-Seidel法となる．

> AA:=Matrix([[5,1,1,1],[1,3,1,1],[1,-2,-9,1],[1,3,-2,5]]):

b:=Vector([-6,2,-7,3]): n:=4;

x0:=[0,0,0,0]: x1:=[0,0,0,0]:

for iter from 1 to 20 do

for i from 1 to n do

x1[i]:=b[i];

for j from 1 to n do

x1[i]:=x1[i]-AA[i,j]*x0[j];

end do:

x1[i]:=x1[i]+AA[i,i]*x0[i];

x1[i]:=x1[i]/AA[i,i];

end do:

x0:=evalf(x1);

print(iter,x0);

end do:

4

1, [-1.200000000, 0.6666666667, 0.7777777778, 0.6000000000]

2, [-1.608888889, 0.6074074073, 0.5629629630, 0.7511111112]

3, [-1.584296296, 0.7649382717, 0.5474897119, 0.7825185186]

4, [-1.618989300, 0.7514293553, 0.5187050756, 0.6768921810]

5, [-1.589405322, 0.8077973477, 0.5061160189, 0.6804222770]

6, [-1.598867129, 0.8009556753, 0.4972691400, 0.6356490634]

7, [-1.586774776, 0.8219829753, 0.4927633981, 0.6381076766]

8, [-1.590570810, 0.8186345670, 0.4897074389, 0.6212705292]

9, [-1.585922507, 0.8265309473, 0.4881589539, 0.6228163974]

10, [-1.587501260, 0.8249823853, 0.4870924439, 0.6165295146]

11, [-1.585720869, 0.8279597673, 0.4865626093, 0.6173477984]

12, [-1.586374035, 0.8272701537, 0.4861897104, 0.6149933572]

13, [-1.585690644, 0.8283969890, 0.4860087794, 0.6153885990]

14, [-1.585958873, 0.8280977553, 0.4858782197, 0.6145034472]

15, [-1.585695884, 0.8285257353, 0.4858165626, 0.6146844092]

16, [-1.585805341, 0.8283983040, 0.4857707838, 0.6143503606]

17, [-1.585703890, 0.8285613990, 0.4857498236, 0.6144303994]

18, [-1.585748324, 0.8285078890, 0.4857337457, 0.6143038680]

19, [-1.585709101, 0.8285702367, 0.4857266407, 0.6143384296]

20, [-1.585727061, 0.8285480103, 0.4857209840, 0.6142903344]

4.7 課題

1. 後退代入法で解を求めよ．(2005年度期末類題)

x+ 4y − 3z = 1
−6y + 4z = 1

− 5
3z = 1

3

2. 次の行列 Aを LU分解せよ．

> A:=Matrix([[1,4,3],[1,-2,1],[2,-2,-1]]); 1 4 3
1 −2 1
2 −2 −1


3. 次の連立方程式の係数行列を LU分解し，上・下三角行列を求めよ．さらに連立方程式の解を求めよ.(2005年

度期末試験) 
x1 + 3x2 + 4x3 + 3x4

−2x1 + 5x2 + 3x3 − 3x4

x1 + 3x2 − 2x3 + 3x4

3x1 − 2x2 + x3 + 4x4

 =


1
4
−2
3


4. Jacobi法のプログラムを参照して Gauss-Seidel法のプログラムを作れ．Jacobi法と収束性を比べよ．

5. 次の連立方程式の解を求めよ．ただし，pivot操作が必要となる．

> with(LinearAlgebra):

A:=Matrix([[3,2,2,1],[3,2,3,1],[1,-2,-3,1],[5,3,-2,5]]):

X:=Vector([w,x,y,z]):

b:=Vector([-6,2,-9,2]):

A.X=b; 
3w + 2x+ 2 y + z
3w + 2x+ 3 y + z
w − 2x− 3 y + z

5w + 3x− 2 y + 5 z

 =


−6
2
−9
2


6. (おまけ）pivot操作を含めた LU分解のプログラムを作成せよ．上の問題を解き，その L, U行列および L−1.b

ベクトルを求めよ．

4.8 解答例

4. Jacobi法のプログラムを参照して Gauss-Seidel法のプログラムを作れ．Jacobi法と収束性を比べよ．

#Gauss-Seidel

AA:=Matrix([[5,1,1,1],[1,3,1,1],[1,-2,-9,1],[1,3,-2,5]]):

b:=Vector([-6,2,-7,3]):

n:=4;

x0:=[0,0,0,0]:

x1:=[0,0,0,0]:

for iter from 1 to 20 do

for i from 1 to n do

x1[i]:=b[i];

for j from 1 to n do

x1[i]:=x1[i]-AA[i,j]*x0[j];

end do:

x1[i]:=x1[i]+AA[i,i]*x0[i];

x1[i]:=x1[i]/AA[i,i];

x0:=evalf(x1); #change here from ...

end do:

print(iter,x0);

end do:

4

1, [-1.200000000, 1.066666667, 0.4074074073, 0.3629629628]

2, [-1.567407407, 0.9323456790, 0.4367626887, 0.5287791494]

3, [-1.579577503, 0.8713452217, 0.4673901337, 0.5800644210]

4, [-1.583759955, 0.8454351333, 0.4783815777, 0.6008435420]

5, [-1.584932051, 0.8352356437, 0.4828266893, 0.6089756998]

6, [-1.585407607, 0.8312017393, 0.4845738460, 0.6121900162]

7, [-1.585593120, 0.8296097527, 0.4852641546, 0.6134584342]

8, [-1.585666468, 0.8289812930, 0.4855365978, 0.6139591570]

9, [-1.585695410, 0.8287332183, 0.4856441456, 0.6141568092]

10, [-1.585706835, 0.8286352933, 0.4856865986, 0.6142348304]

11, [-1.585711344, 0.8285966383, 0.4857033566, 0.6142656284]

12, [-1.585713125, 0.8285813800, 0.4857099714, 0.6142777856]

13, [-1.585713827, 0.8285753567, 0.4857125829, 0.6142825846]

14, [-1.585714105, 0.8285729793, 0.4857136134, 0.6142844788]

15, [-1.585714214, 0.8285720407, 0.4857140204, 0.6142852266]

16, [-1.585714258, 0.8285716703, 0.4857141809, 0.6142855218]

17, [-1.585714275, 0.8285715240, 0.4857142443, 0.6142856384]

18, [-1.585714281, 0.8285714660, 0.4857142694, 0.6142856844]

19, [-1.585714284, 0.8285714433, 0.4857142792, 0.6142857024]

20, [-1.585714285, 0.8285714343, 0.4857142831, 0.6142857096]

5.1 固有値

Aを対称正方行列，xをベクトルとしたときに，

Ax = λx (5.1)

の解，λを固有値，xを固有ベクトルという．xがゼロベクトルではない意味のある解は特性方程式 det(A−λE) = 0

が成り立つときにのみ得られる．

まずMapleで特性方程式を解いてみる．

> restart;

> with(LinearAlgebra):with(plots):with(plottools):

> A:=Matrix(1..2,1..2,[[3,2/3],[2/3,2]]);

A :=

[
3 2/3
2/3 2

]

> EE:=Matrix([[1,0],[0,1]]):

A-lambda.EE;

> eq2:=Determinant(A-lambda.EE);

eq2 :=
50

9
− 5λ+ λ2

> solve(eq2=0,lambda);

10/3, 5/3

固有値を求めるコマンド Eigenvectorsを適用すると，固有値と固有ベクトルが求まる．ここで，固有ベクトルは行

列の列 (Column)ベクトルに入っている．

> lambda,V:=Eigenvectors(A);

λ, V :=

[
10/3
5/3

]
,

[
2 −1/2
1 1

]
得られた固有ベクトルは規格化されているわけではない．

行列の列を取り出すコマンド Columnを用いて，方程式 (5.1)が成り立っていることを確認する．

> lambda[1].Column(V,1)=A.Column(V,1);[
20/3
10/3

]
=

[
20/3
10/3

]
一般的な規格化は，コマンド Normalize(vector,Euclidean)によっておこなう．

> Normalize(Column(v,1),Euclidean);

5.2 固有値の幾何学的意味

次に，固有値の幾何学的な意味を 2次元行列で確認しておこう．ある点 x0 に対称正方行列 Aを作用すると，x1 に

移動する．これを原点を中心とする円上の点に次々に作用させ，移動前後の点を結ぶ．

> restart;

with(LinearAlgebra):with(plots):with(plottools):

A:=Matrix(1..2,1..2,[[3,2/3],[2/3,2]]):

> N:=30:p1:=[]:l1:=[]:

for k from 0 to N-1 do

x0:=Vector([sin(2*Pi*k/N),cos(2*Pi*k/N)]);

x1:=MatrixVectorMultiply(A,x0);

p1:=[op(p1),pointplot({x0,x1})];

l1:=[op(l1),line( evalf(convert(x0,list)),evalf(convert(x1,list)) )];

end do:

> n:=4;

display(p1,l1,view=[-n..n,-n..n]);

x

0-2-4

真ん中の円状の領域が，外側の楕円状の領域に写像されている様子が示されている．この図の中に固有ベクトル，

固有値が隠れている．どこかわかる？

5.3 Googleのページランク

多くの良質なページからリンクされているページはやはり良質なページである

Googleの page rankは上のような非常に単純な仮定から成り立っている．ページランクを実際に求めよう．つぎの

ようなリンクが張られたページを考える．

計算手順は以下の通り*1．

1. リンクを再現する隣接行列を作る．ページに番号をつけて，その間が結ばれている i-j要素を 1，そうでない要

素を 0とする．

2. 隣接行列を転置する

3. 列ベクトルの総和が 1となるように規格化する．

4. こうして得られた推移確率行列の最大固有値に属する固有ベクトルを求め，適当に規格化する．

5.3.1 課題

1. 上記手順を参考にして，Mapleでページランクを求めよ．

2. このような問題ではすべての固有値・固有ベクトルを求める必要はなく，最大の固有値を示す固有ベクトルを

*1 詳しくは http://www.kusastro.kyoto-u.ac.jp/~baba/wais/pagerank.htmlを参照せよ．

求めるだけでよい．初期ベクトルを適当に決めて，何度も推移確率行列を掛ける反復法でページランクを求

めよ．

隣接行列

A1 :=



1 2 3 4 5 6 7
1 0 1 1 1 1 0 1
2 1 0 0 0 0 0 0
3
4
5
6
7


転置行列

Transpose(A1 ) :=




規格化 


遷移 

0 1 1/2 0 1/4 1/2 0
1/5 0 1/2 1/3 0 0 0
1/5 0 0 1/3 1/4 0 0
1/5 0 0 0 1/4 0 0
1/5 0 0 1/3 0 1/2 1
0 0 0 0 1/4 0 0
1/5 0 0 0 0 0 0





1/7
1/7
1/7
1/7
1/7
1/7
1/7


=




=



0.32
0.15
0.11
0.06
0.29
0.04
0.03



5.4 累乗 (べき乗)法により最大固有値が求まる原理

累乗 (べき乗)法は，最大固有値とその固有ベクトルを効率的に見つける算法である．すこし，固有値について復習

しておく．正方行列 Aに対して，
Ax = λx (5.2)

の解 λを固有値，xを固有ベクトルという．λは，

det(A− λE) = 0 (5.3)

として求まる永年方程式の解である．

では，なぜ適当な初期ベクトル x0 から始めて，反復

xk+1 = Axk (5.4)

を繰り返すと，Aの絶対値最大の固有値に属する固有ベクトルに近づいていくのかを見ておこう．

すべての固有値がお互いに異なる場合を考える．今，行列の固有値を絶対値の大きなもの順に並べて，|λ1| > |λ2| >
· · · > |λn|とし，対応する長さを 1に規格化した固有ベクトルを x1, x2, . . . , xn とする．初期ベクトルは固有ベクト

ルの線形結合で表わせて，
X0 = c1x1 + c2x2 + · · ·+ cnxn (5.5)

となるとする．これに行列 Aを N 回掛けると，

ANX0 = c1λ
N
1 x1 + c2λ

N
2 x2 + · · ·+ cnλ

N
n xn (5.6)

となる．これを変形すると，

ANX0 = XN = c1λ
N
1

{
x1 +

c2
c1

(
λ2

λ1

)N

x2 + · · ·+ cn
c1

(
λn

λ1

)N

xn

}
(5.7)

となる．|λ1| > |λi|(i ≥ 2)だから括弧の中は x1 だけが生き残る．

こうして最大固有値に属する固有ベクトルが，反復計算を繰り返すだけで求められる．

5.5 Jacobi回転による固有値の求め方

固有値を求める手法として，永年方程式を解くというやり方は回りくどすぎる．少し古めかしいが非対角要素を 0

にする回転行列を反復的に作用させる Jacobi(ヤコビ) 法を紹介する．現在認められている最適の方策は，ハウスホ

ルダー (Householder)変換で行列を単純な三重対角化行列に変形してから，反復法で解を追い込んでいくやり方であ

る．Jacobi法は，Householder法ほど万能ではないが，10次程度までの行列には今でも役に立つ．

5.5.1 Mapleでみる回転行列

行列の軸回転の復習をする．対称行列 B に回転行列 U を作用すると

B.U =

(
a1 1 a1 2

a2 1(= a1 2) a2 2

)(
cos(θ) − sin(θ)
sin(θ) cos(θ)

)
(5.8)

となる．回転行列を 4x4の行列に
U tBU (5.9)

と作用させたときの各要素の様子を以下に示した．

> restart:

> n:=4:

> with(LinearAlgebra):

> B:=Matrix(n,n,shape=symmetric,symbol=a);

B :=


a1, 1 a1, 2 a1, 3 a1, 4
a1, 2 a2, 2 a2, 3 a2, 4
a1, 3 a2, 3 a3, 3 a3, 4
a1, 4 a2, 4 a3, 4 a4, 4



> U:=Matrix(n,n,[[c,-s,0,0],[s,c,0,0],[0,0,1,0],[0,0,0,1]]);

#U:=Matrix(n,n,[[c,-s],[s,c]]);

U :=


c −s 0 0
s c 0 0
0 0 1 0
0 0 0 1



>TT:=Transpose(U).B.U;

TT := (5.10)

[ (c a1, 1 + s a1, 2) c+ (c a1, 2 + s a2, 2) s , −(c a1, 1 + s a1, 2) s+ (c a1, 2 + s a2, 2) c ,
c a1, 3 + s a2, 3 , c a1, 4 + s a2, 4 ]

[ (−s a1, 1 + c a1, 2) c+ (−s a1, 2 + c a2, 2) s , −(−s a1, 1 + c a1, 2) s+ (−s a1, 2 + c a2, 2) c ,
−s a1, 3 + c a2, 3 , −s a1, 4 + c a2, 4 ]

[c a1, 3 + s a2, 3 , −s a1, 3 + c a2, 3 , a3, 3 , a3, 4]
[c a1, 4 + s a2, 4 , −s a1, 4 + c a2, 4 , a3, 4 , a4, 4]

>expand(TT[1,1]);

expand(TT[2,2]);

expand(TT[1,2]);

expand(TT[2,1]);

c2 a1, 1 + 2 c s a1, 2 + s2 a2, 2

s2 a1, 1 − 2 c s a1, 2 + c2 a2, 2

−s c a1, 1 − s2 a1, 2 + c2 a1, 2 + c s a2, 2

−s c a1, 1 − s2 a1, 2 + c2 a1, 2 + c s a2, 2

この非対角要素を 0にする θ は以下のように求まる．

このとき注目している i, j = 1, 2以外の要素も変化する．

>expand(TT[3,1]);

expand(TT[3,2]);

c a1, 3 + s a2, 3

−s a1, 3 + c a2, 3

これによって一旦 0になった要素も値を持つが，なんども繰り返すことによって，徐々に 0へ近づいていく．

5.6 Jacobi法による固有値を求める Cコード

以下にはヤコビ法を用いた固有値と固有ベクトルを求めるコードを示した．結果は，固有値とそれに対応する規格

化された固有ベクトルが縦 (column)ベクトルで表示される．

リスト: ヤコビ法．

#include <stdio.h>

#include <math.h>

#define M 10

void PrintMatrix(double a[M][M], int n);

int main(void){

double a[M][M],v[M][M];

double eps=0.0001,div,r,t,s,c,apj,aqj,aip,aiq,vip,viq;

int i,j,n,iter,count,iterMax=1000000,p,q;

scanf("%d",&n);

for(i=1;i<=n;i++){

for(j=1;j<=n;j++) scanf("%lf",&a[i][j]);

}

PrintMatrix(a,n);

for(i=1;i<=n;i++){

for(j=1;j<=n;j++) v[i][j]=0.;

v[i][i]=1.;

}

for(iter=1;iter<=iterMax;iter++){

count=0;

for(p=1;p<=n-1;p++){

for(q=p+1;q<=n;q++){

if(fabs(a[p][q])<eps) continue;

count++;

div=a[p][p]-a[q][q];

if (div != 0.0){

r=2.0*a[p][q]/div;

t=0.5*atan(r);

} else {

t=0.78539818;

}

s=sin(t);

c=cos(t);

for(j=1;j<=n;j++){

apj=a[p][j];

aqj=a[q][j];

a[p][j]=apj*c+aqj*s;

a[q][j]=-apj*s+aqj*c;

}

for(i=1;i<=n;i++){

aip=a[i][p];

aiq=a[i][q];

a[i][p]=aip*c+aiq*s;

a[i][q]=-aip*s+aiq*c;

vip=v[i][p];

viq=v[i][q];

v[i][p]=vip*c+viq*s;

v[i][q]=-vip*s+viq*c;

}

printf("p,q=%3d,%3d\n",p,q);

PrintMatrix(a,n);

}

}

if (count==0) break;

}

printf("Eigen values:\n");

for(i=1;i<=n;i++) printf("%6.2f",a[i][i]);

printf("\nEigen vectors:\n");

PrintMatrix(v,n);

return 0;

}

void PrintMatrix(double a[M][M], int n){

int i,j;

for(i=1;i<=n;i++){

for(j=1;j<=n;j++) printf("%6.2f",a[i][j]);

printf("\n");

}

printf("\n");

}

リスト: ヤコビ法の計算結果．

[BobsNewPBG4:~/NumRecipe/chap8] bob% cat input.txt

4

5 4 1 1

4 5 1 1

1 1 4 2

1 1 2 4

BobsNewPBG4:~/NumRecipe/chap8] bob% Jacobi2<input.txt

5.00 4.00 1.00 1.00

4.00 5.00 1.00 1.00

1.00 1.00 4.00 2.00

1.00 1.00 2.00 4.00

p,q= 1, 2

9.00 -0.00 1.41 1.41

-0.00 1.00 -0.00 -0.00

1.41 -0.00 4.00 2.00

1.41 -0.00 2.00 4.00

p,q= 1, 3

9.37 -0.00 -0.00 1.88

-0.00 1.00 0.00 -0.00

-0.00 0.00 3.63 1.57

1.88 -0.00 1.57 4.00

p,q= 1, 4

9.96 -0.00 0.47 -0.00

-0.00 1.00 0.00 0.00

0.47 0.00 3.63 1.50

0.00 0.00 1.50 3.41

...<中略>...

Eigen values:

10.00 1.00 5.00 2.00

Eigen vectors:

0.63 -0.71 -0.32 0.00

0.63 0.71 -0.32 0.00

0.32 0.00 0.63 -0.71

0.32 0.00 0.63 0.71

5.7 数値計算ライブラリーについて

一般の数値計算ライブラリーについては，時間の関係で講義ではその能力を紹介するにとどめる．昔の演習で詳し

く取り上げていたので，研究や今後のために必要と思うときは，テキストを取りにおいで．

行列の計算は，数値計算の中でも特に利用する機会が多く，また，律速ルーチンとなる可能性が高い．そこで，古

くから行列計算の高速ルーチンが開発されてきた．なかでも BLASと LAPACKはフリーながら非常に高速である．

前回に示した，逆行列を求める単純な LU分解法を C言語でコーディングしたものと，LAPACKのルーチンを比

べた場合，1000次元の行列で計測すると

> 1000 [dim] 2.5200 [sec] #BOB

> 1000 [dim] 0.4700 [sec] #LAPACK

となった．2006 年に初めてこの計算に用いた PC は MacBook(2GHz, Intel Core Duo) であるが，この計算での

0.47秒は 1.4GFLOPに相当する．07年のMacBook(2GHz, Intel Core 2 Duo)ではさらに速くなって

bob% gcc -O3 bob.c -o bob

bob% ./bob

1000

1000 [dim] 1.7543 [sec] #BOB

bob% gcc -O3 lapack.c -llapack -lblas -o lapack

bob% ./lapack

1000

1000 [dim] 0.1893 [sec] #LAPACK

で，3.5GFLOPSが出ている．今 (2016年)は，MacBookAir(2.2GHz, Intel Core i7)で... top500.orgが毎年２回

High Performance Computerのランクを発表している．今は，Top1は 100PFlopsであるが，初回の 1994年 6月

の 500位は 0.4GFlopsで，今の laptopがはるかに凌いでいる．まさに laptopスパコンの時代なんですよ．

ライブラリーは世界中の計算機屋さんがよってたかって検証しているので，バグがほとんど無く，また，高速であ

る．初学者はライブラリーを使うべきである．ただし，下のサンプルプログラムの行列生成の違いのように，ブラッ

クボックス化すると思わぬ間違い（ここでは Fortranと Cでの行列の並び順の違いが原因)をしでかすことがあるの

で，プログラムに組み込む前に必ず小さい次元 (サンプルコード)で検証しておくこと*2．

添付のコードはちょっと長いが時間があればフォローせよ．コンパイルは，OSXでは

> gcc -O3 -UPRINT lapack.c -llapack -lblas

とすればできる．linuxでは LAPACK, BLASがインストールされていれば，

> #include <vecLib/vecLib.h>

をコメントアウトして，

> gcc -O3 -DPRINT lapack.c -L/usr/local/lib64 -llapack -lblas -lg2c

などとすればコンパイルできるはず．

■リスト: 西谷製 lazy逆行列計算プログラム 　

*2 少し前 (2002年ごろ)GotoBLASが開発されて，性能が 10%ほども上がった

#include <stdio.h>

#include <stdlib.h>

#include <math.h>

#include <time.h>

//#undef PRINT

//#define PRINT

void printMatrix(double *a, double *b, long n);

int MatrixInverse(double *a, double *b, long n);

int main(void){

clock_t start, end;

int i,j;

long n;

double *a,*b;

scanf("%ld",&n);

a=(double *)malloc(n*n*sizeof(double));

b=(double *)malloc(n*sizeof(double));

for (i=0;i<n;i++){

for (j=0;j<n;j++){

a[i*n+j]= 2*(double) random() / RAND_MAX - 1.0;

}

}

for (i=0;i<n;i++){

b[i]= 2*(double) random() / RAND_MAX - 1.0;

}

printMatrix(a,b,n);

start = clock();

MatrixInverse(a,b,n);

end = clock();

printf("%5d [dim] %10.4f [sec] #BOB\n",

n,(double)(end-start)/CLOCKS_PER_SEC);

printMatrix(a,b,n);

free(a);

free(b);

return 0;

}

int MatrixInverse(double *a, double *b, long n){

double *x;

double pvt=0.00005,am;

int i,j,k;

x=(double *)malloc(n*sizeof(double));

for(i=0;i<n-1;i++){

if(fabs(a[i*n+i])<pvt){

printf("Pivot %3d=%10.5f is too small.\n",i,a[i*n+i]);

return 1;

}

for(j=i+1;j<n;j++){

am=a[j*n+i]/a[i*n+i];

for(k=0;k<n;k++) a[j*n+k]-=am*a[i*n+k];

b[j]-=am*b[i];

}

}

//Backward substitution

for(j=n-1;j>=0;j--){

x[j]=b[j];

for(k=j+1;k<n;k++){

x[j]-=a[j*n+k]*x[k];

}

b[j]=x[j]/=a[j*n+j];

}

free(x);

return 0;

}

void printMatrix(double *a, double *b, long n){

int i,j;

#ifdef PRINT

printf("\n");

for (i=0;i<n;i++){

for (j=0;j<n;j++){

printf("%10.5f",a[i*n+j]);

}

printf(":%10.5f",b[i]);

printf("\n");

}

printf("\n");

#endif

return;

}

■リスト : LAPACK謹製 smart逆行列計算プログラム 　

#include <stdio.h>

#include <stdlib.h>

#include <math.h>

#include <time.h>

//#define PRINT

//#undef PRINT

void printMatrix(double *a, double *b, long n);

int main(void){

clock_t start, end;

int i,j;

double *a,*b;

long n,nrhs=1, lda,ldb, info, *ipiv;

scanf("%ld",&n);

a=(double *)malloc(n*n*sizeof(double));

b=(double *)malloc(n*sizeof(double));

lda=ldb=n;

ipiv=(long *)malloc(n*sizeof(long));

for (i=0;i<n;i++){

for (j=0;j<n;j++){

a[j*n+i]= 2*(double) random() / RAND_MAX - 1.0;

}

}

for (i=0;i<n;i++){

b[i]= 2*(double) random() / RAND_MAX - 1.0;

}

printMatrix(a,b,n);

start = clock();

dgesv_(&n, &nrhs, a, &lda, ipiv, b, &ldb, &info);

end = clock();

printf("%5d [dim] %10.4f [sec] #LAPACK\n",

n, (double)(end-start)/CLOCKS_PER_SEC);

printMatrix(a,b,n);

free(a);

free(b);

free(ipiv);

return 0;

}

void printMatrix(double *a, double *b, long n){

int i,j;

#ifdef PRINT

printf("\n");

for (i=0;i<n;i++){

for (j=0;j<n;j++){

printf("%10.5f",a[i*n+j]);

}

printf(":%10.5f",b[i]);

printf("\n");

}

printf("\n");

#endif

return;

}

5.8 課題

1. 4x4の行列を適当に作り，Mapleで固有値を求めよ．求め方はマニュアルを参照せよ．

2. Jacobi法によって固有値を求めよ．

3. LAPACKに含まれている dsyev関数を用いて実対称行列の固有値を求めよ．(演習で詳しく取り上げている．

研究や今後のために必要と思うときは，テキストを取りにおいで)



53

第 6章

補間 (interpolation)と数値積分 (Integral)

6.1 概要:補間と近似

単純な 2次元データについて補間と近似を考える．補間はたんに点をつなぐことを，近似はある関数にできるだけ

近くなるようにフィットすることを言う．補間は Illustratorなどのドロー系ツールで曲線を引くときの，ベジエやス

プライン補間の基本となる．本章では補間とそれに密接に関連した積分について述べる．

表 6.1 補間と近似の模式図．

補間 近似

6.2 多項式補間 (polynomial interpolation)

データを単純に多項式で補間する方法を先ず示そう．N + 1点を N次の多項式でつなぐ．この場合の補間関数は，

F (x) =
N∑
i=0

aix
i = a0 + a1x+ a2x

2 + · · ·+ aNxN

である．データの点を (xi, yi), i = 0..N とすると

a0 + a1x0 + a2x
2
0 + · · ·+ aNxN

0 = y0
a0 + a1x1 + a2x

2
1 + · · ·+ aNxN

1 = y1
...

a0 + a1xN + a2x
2
N + · · ·+ aNxN

N = yN

が，係数 ai を未知数と見なした線形の連立方程式となっている．係数行列は

A =


1 x0 x2

0 · · · xN
0

1 x1 x2
1 · · · xN

1
...

...
1 xN x2

N · · · xN
N


となる．ai と yi をそれぞれベクトルとみなすと

により未知数ベクトル ai が求まる．これは単純に，前に紹介した Gauss の消去法や LU 分解で解ける．

6.2.1 Mapleによる多項式補間の実例

> restart; X:=[0,1,2,3]: Y:=[1,2,3,-2]:

> with(LinearAlgebra):

> list1:=[X,Y];

list1 := [[0, 1, 2, 3], [1, 2, 3,−2]]

> with(plots):

l1p:=listplot(Transpose(Matrix(list1))):

display(l1p);

> A:=Matrix(4,4):

for i from 1 to 4 do

for j from 1 to 4 do

A[i,j]:=X[i]^(j-1);

end do;

end do:

A; 
1 0 0 0
1 1 1 1
1 2 4 8
1 3 9 27



> a1:=MatrixInverse(A).Vector(Y);

a1 :=


1
−1
3
−1



> f1:=unapply(add(a1[ii]*x^(ii-1),ii=1..4),x);

f1 := x 7→ 1− x+ 3x2 − x3

> f1p:=plot(f1(x),x=0..3):

l1p:=listplot(Transpose(Matrix(list1))):

display(f1p,l1p);

6.3 Lagrange(ラグランジュ) の内挿公式

多項式補間は手続きが簡単であるため，計算間違いが少なく，プログラムとして組むのに適している．しかし，あ

まり”みとうし”のよい方法とはいえない．その点，Lagrange(ラグランジュ)の内挿公式は見通しがよい．これは

F (x) =

N∑
k=0

∏
j ̸=k

(x− xj)∏
j ̸=k

(xk − xj)
yk =

N∑
k=0

(x− x0)(x− x1) · · · (x− xN )
(x− xk)

(xk − x0)(xk − x1) · · · (xk − xN )
(xk − xk)

yk

と表わされる．数学的に 2つ目の表記は間違っているが，先に割り算を実行すると読み取って欲しい．これは一見複

雑に見えるが，単純な発想から出発している．求めたい関数 F (x)を

F (x) = y0L0(x) + y1L1(x) + y2L2(x)

とすると
L0(x0) = 1 L0(x1) = 0 L0(x2) = 0
L1(x0) = 0 L1(x1) = 1 L1(x2) = 0
L2(x0) = 0 L2(x1) = 0 L2(x2) = 1

となるように関数 Li(x)を決めればよい．これを以下のようにとれば Lagrangeの内挿公式となる．

6.4 Newton(ニュートン) の差分商公式

もう一つ有名な Newton(ニュートン) の内挿公式は，

F (x) = F (x0) + (x− x0)f1⌊x0, x1⌋+ (x− x0)(x− x1)f2⌊x0, x1, x2⌋+

· · ·+
n−1∏
i=0

(x− xi) fn⌊x0, x1, · · · , xn⌋

となる．ここで fi⌊ ⌋ は次のような関数を意味していて，

f1⌊x0, x1⌋ =
y1 − y0
x1 − x0

f2⌊x0, x1, x2⌋ =
f1⌊x1, x2⌋ − f1⌊x0, x1⌋

x2 − x0
...

fn⌊x0, x1, · · · , xn⌋ =
fn−1⌊x1, x2 · · · , xn⌋ − fn−1⌊x0, x1, · · · , xn−1⌋

xn − x0

差分商と呼ばれる．得られた多項式は，Lagrange の内挿公式で得られたものと当然一致する．Newtonの内挿公式

の利点は，新たなデータ点が増えたときに，新たな項を加えるだけで，内挿式が得られる点である．

6.4.1 Newton補間と多項式補間の一致の検証

関数 F (x)を xの多項式として展開．その時の，係数の取るべき値と，差分商で得られる値が一致．

> restart: F:=x->f0+(x-x0)*f1p+(x-x0)*(x-x1)*f2p;

F := x 7→ f0 + (x− x0)f1p+ (x− x0)(x− x1)f2p

> F(x1);

sf1p:=solve(F(x1)=f1,f1p);

f0 + (x1− x0)f1p

sf1p :=
f0− f1

−x1 + x0

f20の取るべき値の導出

> sf2p:=solve(F(x2)=f2,f2p);

fac_f2p:=factor(subs(f1p=sf1p,sf2p));

sf2p := −f0 + f1p x2− f1p x0− f2

(−x2 + x0)(−x2 + x1)

fac f2p :=
f0x1− x2 f0 + x2 f1− x0 f1− f2x1 + f2x0

(−x1 + x0)(−x2 + x0)(−x2 + x1)

ニュートンの差分商公式を変形

> ff11:=(f0-f1)/(x0-x1);

ff12:=(f1-f2)/(x1-x2);

ff2:=(ff11-ff12)/(x0-x2);

fac_newton:=factor(ff2);

ff11 :=
f0− f1

−x1 + x0

ff12 :=
f1− f2

−x2 + x1

ff2 :=

f0− f1

−x1 + x0
− f1− f2

−x2 + x1
−x2 + x0

fac newton :=
f0x1− x2 f0 + x2 f1− x0 f1− f2x1 + f2x0

(−x1 + x0)(−x2 + x0)(−x2 + x1)

二式が等しいかどうかを evalbで判定

> evalb(fac_f2p=fac_newton);

true

6.5 数値積分 (Numerical integration)

積分，

I =

∫ b

a

f(x)dx

を求めよう．1次元の数値積分法では連続した領域を細かい短冊に分けて，それぞれの面積を寄せ集めることに相当

する．分点の数を N とすると，

xi = a+
b− a

N
i = a+ h× i

h =
b− a

N

ととれる．そうすると，もっとも単純には，

IN =

{
N−1∑
i=0

f(xi)

}
h =

{
N−1∑
i=0

f(a+ i× h)

}
h

となる．

表 6.2 数値積分の模式図．

6.5.1 中点則 (midpoint rule)

中点法 (midpoint rule) は，短冊を左端から書くのではなく，真ん中から書くことに対応し，

IN =

{
N−1∑
i=0

f

(
a+

(
i+

1

2

)
× h

)}
h

となる．

6.5.2 台形則 (trapezoidal rule)

さらに短冊の上側を斜めにして，短冊を台形にすれば精度が上がりそうに思う．その場合は，短冊一枚の面積 Si

は，

Si =
f(xi) + f(xi+1)

2
h

で求まる．これを端から端まで加えあわせると，

iN =
N−1∑
i=0

Si = h

{
1

2
f(x0) +

N−1∑
i=1

f(xi) +
1

2
f (xN )

}

が得られる．

6.5.3 Simpson(シンプソン)則

Simpson(シンプソン) 則では，短冊を 2次関数，

f(x) = ax2 + bx+ c

で近似することに対応する．こうすると，

Si =

∫ xi+1

xi

f(x) dx =

∫ xi+1

xi

(ax2 + bx+ c) dx

h

6

{
f(xi) + 4f

(
xi +

h

2

)
+ f(xi + h)

}
となる．これより，

IN =
h

6

{
f (x0) + 4

N−1∑
i=0

f

(
xi +

h

2

)
+ 2

N−1∑
i=1

f (xi) + f (xN )

}

として計算できる．ただし，関数値を計算する点の数は台形則などの倍となっている．

教科書によっては，分割数 N を偶数にして，点を偶数番目 (even) と奇数番目 (odd) に分けて，

IN =
h

3

{
f (x0) + 4

N−2∑
i=even

f

(
xi +

h

2

)
+ 2

N−1∑
i=odd

f (xi) + f (xN )

}

としている記述があるが，同じ計算になるので誤解せぬよう．

6.6 数値積分のコード

次の積分を例に，Mapleのコードを示す． ∫ 1

0

4

1 + x2
dx

先ずは問題が与えられたらできるだけMapleで解いてしまう．答えをあらかじめ知っておくと間違いを見つけるのが

容易．プロットしてみる．

> restart;

f1:=x->4/(1+x^2);

plot(f1(x),x=0..5);

f1 := x 7→ 4

1 + x2

Mapleで解いてみる．

>int(f1(x),x=0..1);

π

えっと思うかも知れないが，

>int(1/(1+x^2),x);

arctan(x)

となるので，納得できるでしょう．

具体的にMapleでコードを示す．先ずは初期設定．

>N:=8: x0:=0: xn:=1: Digits:=20:

■Midpoint rule(中点法) 　

> h:=(xn-x0)/N: S:=0:

for i from 0 to N-1 do

xi:=x0+(i+1/2)*h;

dS:=h*f1(xi);

S:=S+dS;

end do:

evalf(S);

3.1428947295916887799

■Trapezoidal rule(台形公式) 　

> h:=(xn-x0)/N: S:=f1(x0)/2:

for i from 1 to N-1 do

xi:=x0+i*h;

dS:=f1(xi);

S:=S+dS;

end do:

S:=S+f1(xn)/2:

evalf(h*S);

3.1389884944910890093

■Simpson’s rule(シンプソンの公式) 　

> M:=N/2: h:=(xn-x0)/(2*M): Seven:=0: Sodd:=0:

for i from 1 to 2*M-1 by 2 do

xi:=x0+i*h;

Sodd:=Sodd+f1(xi);

end do:

for i from 2 to 2*M-1 by 2 do

xi:=x0+i*h;

Seven:=Seven+f1(xi);

end do:

evalf(h*(f1(x0)+4*Sodd+2*Seven+f1(xn))/3);

3.1415925024587069144

6.7 課題

1. 補間と近似の違いについて，適切な図を描いて説明せよ．

2. 次の 4点

x y

0 1

1 2

2 3

3 -2

を通る多項式を以下のそれぞれの手法で求めよ．(a) 逆行列, (b) ラグランジュ補間, (c) ニュートンの差分商

公式　

3. tan(5◦) = 0.08748866355, tan(10◦) = .1763269807, tan(15◦) = .2679491924の値を用いて，ラグランジュ補

間法により，tan(17◦)の値を推定せよ．(2008年度期末試験）

4. (対数関数のニュートンの差分商補間:2014期末試験，25点)

2を底とする対数関数 (Mapleでは log[2](x))の F (9.2) = 2.219203をニュートンの差分商補間を用いて求め

る．ニュートンの内挿公式は，

F (x) = F (x0) + (x− x0)f1⌊x0, x1⌋+ (x− x0)(x− x1)f2⌊x0, x1, x2⌋+
· · ·+

∏n−1
i=0 (x− xi) fn⌊x0, x1, · · · , xn⌋

である．ここで fi⌊ ⌋ は次のような関数を意味していて，

f1⌊x0, x1⌋ =
y1 − y0
x1 − x0

f2⌊x0, x1, x2⌋ =
f1⌊x1, x2⌋ − f1⌊x0, x1⌋

x2 − x0

...

fn⌊x0, x1, · · · , xn⌋ =
fn−1⌊x1, x2 · · · , xn⌋ − fn−1⌊x0, x1, · · · , xn−1⌋

xn − x0

差分商と呼ばれる．xk = 8.0, 9.0, 10.0, 11.0 をそれぞれ選ぶと，差分商補間のそれぞれの項は以下の通りと

なる．

k xk yk = F0(xk) f1⌊xk, xk+1⌋ f2⌊xk, xk+1, xk+2⌋ f3⌊xk, xk+1, xk+2, xk+3⌋
0 8.0 2.079442

0.117783

1 9.0 2.197225 　　ア　　
0.105360 0.0003955000

2 10.0 2.302585 −0.0050250
0.095310

3 11.0 2.397895

それぞれの項は，例えば，

f2⌊x1, x2, x3⌋ =
0.095310− 0.105360

11.0− 9.0
= −0.0050250

で求められる．ニュートンの差分商の一次多項式は

F (x) = F0(8.0) + (x− x0)f1⌊x1, x0⌋ = 2.079442 + 0.117783(9.2− 8.0) = 2.220782

となる．

（a）差分商補間の表中の開いている箇所 ア　 f2⌊x0, x1, x2⌋ を埋めよ．
（b）ニュートンの二次多項式

F (x) = F (x0) + (x− x0)f1⌊x0, x1⌋+ (x− x0)(x− x1)f2⌊x0, x1, x2⌋

の値を求めよ．

（c）ニュートンの三次多項式の値を求めよ．

ただし，ここでは有効数字 7桁程度はとるように．(E.クライツィグ著「数値解析」(培風館,2003), p.31, 例 4

改)

5. exp(0)=1.0, exp(0.1)=1.1052, exp(0.3)=1.3499の値を用いて，ラグランジュ補間法により，exp(0.2)の値を

推定せよ．(2009年度期末試験）

6. 次の関数

f(x) =
4

1 + x2

を x = 0..1で数値積分する．

（a）N を 2,4,8,…256ととり，N 個の等間隔な区間にわけて中点法で求めよ．(15)

（b）小数点以下 10 桁まで求めた値 3.141592654 との差を dX とする．dX と分割数 N とを両対数プロット

(loglogplot)して比較せよ (10)

(2008年度期末試験）

7. 次の関数

y =
1

1 + x2

を x = 0..1で等間隔に N + 1点とり，N 個の区間にわけて数値積分で求める．N を 2, 4, 8, 16, 32, 64, 128,

256と取ったときの (a)中点法, (b)台形公式, (c)シンプソン公式それぞれの収束性を比較せよ．

ヒント：Maple script にあるそれぞれの数値積分法を関数 (procedure) に直して，for-loop で回せば楽．出来

なければ，一つ一つ手で変えても OK. 両対数プロット (loglogplot) すると見やすい．

6.8 解答例

解答例 [2. ] 少し違うデータの場合の結果を以下に示す．データの組は以下のとおり．

> restart;

X:=[0,1,3,2];

Y:=[1,2,-2,3];

with(LinearAlgebra):

list1:=[X,Y];

X := [0, 1, 3, 2]

Y := [1, 2,−2, 3]

list1 := [[0, 1, 3, 2], [1, 2,−2, 3]]

グラフにすると次の通り．

> with(plots):

l1p:=pointplot(Transpose(Matrix(list1))):

display(l1p);

2次,3次までの関数をそれぞれ F2,F3として，Newtonの差分商公式を使った関数を定義する．

> F2:=y0+(x-x0)*f1_01+(x-x0)*(x-x1)*f2_012;

F3:=y0+(x-x0)*f1_01+(x-x0)*(x-x1)*f2_012+(x-x0)*(x-x1)*(x-x2)*f3_0123;

F2 := y0 + (x− x0) f1 01 + (x− x0) (x− x1) f2 012

F3 := y0 + (x− x0) f1 01 + (x− x0) (x− x1) f2 012

+ (x− x0) (x− x1) (x− x2) f3 0123

差分商 f1, f2, f3 の公式をそのまま書き下す．

> f1_01:=(y1-y0)/(x1-x0);

f1_12:=(y2-y1)/(x2-x1);

f1_23:=(y3-y2)/(x3-x2);

f1 01 :=
y1 − y0

x1 − x0

f1 12 :=
y2 − y1

x2 − x1

f1 23 :=
y3 − y2

x3 − x2

> f2_012:=(f1_12-f1_01)/(x2-x0);

f2_123:=(f1_23-f1_12)/(x3-x1);

f2 012 :=
1

x2 − x0

(
y2 − y1

x2 − x1
− y1 − y0

x1 − x0

)
f2 123 :=

1

x3 − x1

(
y3 − y2

x3 − x2
− y2 − y1

x2 − x1

)

> f3_0123:=(f2_123-f2_012)/(x3-x0);

f3 0123 :=
1

x3 − x0

(
1

x3 − x1

(
y3 − y2

x3 − x2
− y2 − y1

x2 − x1

)
− 1

x2 − x0

(
y2 − y1

x2 − x1
− y1 − y0

x1 − x0

))

こうすると F2 は

F2;

y0 +
(x− x0 ) (y1 − y0 )

x1 − x0
+

(x− x0 ) (x− x1 )

x2 − x0

(
y2 − y1

x2 − x1
− y1 − y0

x1 − x0

)
となる．さて，データを追加．次の——は x0,x1,x2,x3を作る．

> for i from 1 to 4 do

x||(i-1):=X[i];

y||(i-1):=Y[i];

end:

値を入れると F2,F3は簡単な数式となる．

F2;

F3;

1 + x− x (x− 1)

1 + x− x (x− 1)− x (x− 1) (x− 3)

> eq2:=expand(F2);

eq3:=expand(F3);

eq2 := −x2 + 2x+ 1

eq3 := −x3 + 3x2 − x+ 1

これで Newtonの差分商公式による内挿関数が求まっている．プロットすると次の通り．

> with(plots):

l1p:=pointplot(Transpose(Matrix(list1))):

pf2:=plot(eq2,x=0..3,color=blue):

pf3:=plot(eq3,x=0..3):

display(l1p,pf2,pf3);

解答例 [6. ] 関数を定義．

> restart;

f1:=unapply(4/(1+x^2),x);

x → 4

x2 + 1

plotしてみる．

> plot(f1(x),x);

Mapleで答えを確認．

> int(f1(x),x=0..1);

π

中点公式を copyする．幾つかの初期値を設定．

> N:=8: x0:=0: xn:=1: Digits:=20:

> mid:=proc(N)

global x0,xn;

local S,i,dS,h,xi;

h:=(xn-x0)/N:

S:=0:

for i from 0 to N-1 do

xi:=x0+(i+1/2)*h;

dS:=h*f1(xi);

S:=S+dS;

end do:

evalf(S);

end;

resultsという配列を用意して，分割点の数と中点公式で得られた結果をためていく．正しい答え (π)との差を

とるが，あとで loglogplotするので，誤差の絶対値をためておく．

> results:=[];

for i from 1 to 8 do

results:=[op(results),[2^i,evalf(abs(mid(2^i)-Pi))]];

end:

loglogplotの結果．正しい解への収束の様子が表示される．

with(plots):

loglogplot([results]);

第 7章

線形最小 2乗法 (LeastSquareFit)

7.1 Mapleによる最小 2乗法

前章では，データに多項式を完全にフィットする補間についてみた．今回は，近似的にフィットする最小二乗法に

ついて詳しくみていく．図のようなデータに直線をフィットする場合を考えよう．

コマンド leastsquareによる fitting(2変数の例)

> restart: X:=[1,2,3,4]: Y:=[0,5,15,24]:

> with(plots):with(linalg):with(stats):

> l1:=pointplot(transpose([X,Y]),symbolsize=30):

> eq_fit:=fit[leastsquare[[x, y], y = a0+a1*x, {a0,a1}]]([X, Y]);

eq fit := y = −19

2
+

41

5
x

> f1:=unapply(rhs(eq_fit),x);

f1 := x 7→ −19

2
+

41

5
x

> p1:=plot(f1(x),x=0..4):

> display(p1,l1);

7.2 最小 2乗法の原理

もっとも簡単な例で原理を解説する．近似関数として，

F (x) = a0 + a1 x

という直線近似を考える．もっともらしい関数は N 点の測定データとの差 di = F (xi)− yi を最小にすればよさそう

であるが，これはプラスマイナスですぐに消えて不定になる．そこで，

χ2 =

N∑
i

d2i =

N∑
i

(a0 + a1 xi − yi)
2

という関数を考える．この χ2(カイ二乗)関数が，a0, a1 をパラメータとして変えた時に最小となる a0, a1 を求める．

これは，それらの微分がそれぞれ 0となる場合である．これは χ2 の和
∑

(sum)の中身を展開し，

χ2 =

a0, a1 でそれぞれ微分すれば

という a0, a1 を未知変数とする 2 元の連立方程式が得られる．これは前に説明した通り逆行列で解くことがで

きる．

7.3 χ2 の極小値から (2変数の例)

> restart; X:=[1,2,3,4]: Y:=[0,5,15,24]: f1:=x->a0+a1*x:

S:=0:

for i from 1 to 4 do

∂

∂a0
χ2 =

∂

∂a1
χ2 =

S:=S+(f1(X[i])-Y[i])^2;

end do:

> fS:=unapply(S,(a0,a1));

fS := (a0 , a1 ) 7→ (a0 + a1 )
2
+ (a0 + 2 a1 − 5)

2
+ (a0 + 3 a1 − 15)

2
+ (a0 + 4 a1 − 24)

2

> expand(fS(a0,a1));

4 a0 2 + 20 a0 a1 + 30 a1 2 − 88 a0 − 302 a1 + 826

> plot3d(fS(a0,a1),a0=-20..20,a1=0..20);

> eqs:={diff(expand(S),a0)=0, diff(expand(S),a1)=0};

eqs := {8 a0 + 20 a1 − 88 = 0, 20 a0 + 60 a1 − 302 = 0}

> solve(eqs,{a0,a1}); {
a0 = −19

2
, a1 =

41

5

}

7.4 正規方程式 (Normal Equations)による解

より一般的な場合の最小二乗法の解法を説明する．先程の例では 1次の多項式を近似関数とした．これをより一般

的な関数，例えば，sin, cos, tan, exp, sinhなどとする．これを線形につないだ関数を

F (x) = a0 sin (x) + a1 cos (x) + a2 exp (−x) + a3 sinh (x) + · · · =
M∑
k=1

akXk (x)

ととる．実際には，Xk(x) はモデルや，多項式の高次項など論拠のある関数列をとる．これらを基底関数 (base

functions)と呼ぶ．ここで線形といっているのは，パラメータ ak について線形という意味である．このような，よ

り一般的な基底関数を使っても，χ2 関数は

χ2 =
N∑
i=1

(F (xi)− yi)
2
=

N∑
i=1

(
M∑
k=1

akXk (xi)− yi

)2

と求めることができる．この関数を，ak を変数とする関数とみなす．この関数が最小値を取るのは，χ2 をM 個の

ak で偏微分した式がすべて 0となる場合である．これを実際に求めてみると，

N∑
i=1

 M∑
j=1

ajXj (xi)− yi

Xk (xi) = 0

となる．ここで，k = 1..M の M 個の連立方程式である．この連立方程式を最小二乗法の正規方程式 (normal

equations)と呼ぶ．

上記の記法のままでは，ややこしいので，行列形式で書き直す．N ×M で，各要素を

Aij = Xj(xi)

とする行列 Aを導入する．この行列は，

A =



X1(x1) X2(x1) · · · XM (x1)
...

... · · ·
...

...
... · · ·

...
...

... · · ·
...

X1(xN ) X2(xN ) · · · XM (xN )


となる．これをデザイン行列と呼ぶ．すると先程の正規方程式は，

At.A.a = At.y

で与えられる．At は行列 Aの転置 (transpose)

At = At
ij = Aji

を意味し，得られた行列は，M ×N である．a, y はそれぞれ，

a =


a1
a2
...

aM

 , y =


y1
y2
...
yN


である．

M = 3, N = 25として行列の次元だけで表現すると，

 · · ·
· · · · · · · · · · · · · · ·

· · ·




...

...
· · · · · · · · ·

...

...




...
...
...

 =

 · · ·
· · · · · · · · · · · · · · ·

· · ·




...

...

...

...

...


となる．これは少しの計算で 3× 3の逆行列を解く問題に変形できる．

7.4.1 Mapleによる具体例

> restart; X:=[1,2,3,4]: Y:=[0,5,15,24]:

f1:=x->a[1]+a[2]*x+a[3]*x^2:

with(LinearAlgebra): Av:=Matrix(1..4,1..3):

ff:=(x,i)->x^(i-1):

for i from 1 to 3 do

for j from 1 to 4 do

Av[j,i]:=ff(X[j],i);

end do;

end do;

Av; 
1 1 1
1 2 4
1 3 9
1 4 16



> Ai:=MatrixInverse(Transpose(Av).Av);

Ai :=


31

4
−27

4

5

4

−27

4

129

20
−5

4
5

4
−5

4

1

4



> b:=Transpose(Av).Vector(Y);

b :=

 44
151
539



> Ai.b; 
−9

2
16

5
1



7.5 特異値分解 (Singular Value Decomposition)による解

正規方程式を解くときには，少し注意が必要である．正規方程式での共分散行列,特異値分解の導出や標準偏差と

の関係は NumRecipeを参照せよ．

> restart; X:=[1,2,3,4]: Y:=[0,5,15,24]: f1:=x->a[1]+a[2]*x+a[3]*x^2:

> with(LinearAlgebra): Av:=Matrix(1..4,1..3):

> ff:=(x,i)->x^(i-1):

for i from 1 to 3 do

for j from 1 to 4 do

Av[j,i]:=ff(X[j],i);

end do;

end do;

Av; 
1 1 1
1 2 4
1 3 9
1 4 16



> U,S,Vt:=evalf(SingularValues(Av,output=[’U’,’S’,’Vt’])):

> DiagonalMatrix(S[1..3],4,3); U.DiagonalMatrix(S[1..3],4,3).Vt:
19.6213640200000015 0 0

0 1.71206987399999999 0
0 0 0.266252879300000022
0 0 0



> iS:=Vector(3):

for i from 1 to 3 do

iS[i]:=1/S[i];

end do:

> DiS:=DiagonalMatrix(iS[1..3],3,4);

DiS :=

 0.05096485642 0 0 0
0 0.5840883104 0 0
0 0 3.755827928 0



> Transpose(Vt).DiS.(Transpose(U).Vector(Y)); −4.50000000198176498
3.20000000035008324
1.00000000040565196



7.6 2次元曲面へのフィット

先程の一般化をより発展させると，3次元 (xi, yi, zi)で提供されるデータへの，2次元平面でのフィットも可能と

なる．2次元の単純な曲面は，方程式を使って，

F (x, y) = a1 + a2 x+ a3 y + a4 xy + a5 x
2 + a6 y

2

となる．デザイン行列の i行目の要素は，
[1, xi, yi, xi yi, x

2
i , y

2
i ]

として，それぞれ求める．このデータの変換の様子をMapleスクリプトで詳しく示した．後は，通常の正規方程式を

解くようにすれば，このデータを近似する曲面を定めるパラメータ a1, a2, · · · , a6 が求まる．最小二乗法はパラメー
タ ak について線形であればよい．

7.6.1 Mapleによる具体例

実際のデータ解析での例．データの座標を x,y,zで用意して，Mapleの埋め込み関数の leastsquareで fitしている．

> with(plots):with(plottools):

z:=[0.000046079702088, 0.000029479057275,

0.000025769637830, 0.000034951410953, 0.000057024385455, 0.000029485453808,

0.000011519913869, 0.000006442404299, 0.000014252898382, 0.000034951410953,

0.000025769637773, 0.000006442404242, 0.000000000000057, 0.000006442404242,

0.000025769637773, 0.000034932221524, 0.000014246501905, 0.000006442404299,

0.000011519913926, 0.000029479057332, 0.000056973214100, 0.000034932221467,

0.000025769637773, 0.000029485453808, 0.000046079702031]:

> x:=[]:

y:=[]:

p1:=2:

for i from -p1 to p1 do

for j from -p1 to p1 do

x:=[op(x),i*0.0005];

y:=[op(y),j*0.0005];

end do;

end do;

> with(LinearAlgebra): p2:=convert(Transpose(Matrix([x,y,z])),listlist):

pp2:=pointplot3d(p2,symbol=circle,symbolsize=30,color=black):

with(stats): data:=[x,y,z]:

fit1:=fit[leastsquare[[t,s,u],

u=a1+a2*t+a3*s+a4*t*s+a5*t^2+a6*s^2,

{a1,a2,a3,a4,a5,a6}]](data);

fit1 := u = −8.657142857× 10−13 − 0.000006396456800 t+ 0.000006396438400 s

−5.459553587 ts+ 25.76962838 t2 + 25.76962835 s2

> f1:=unapply(rhs(fit1),(s,t)):

> pf1:=plot3d(f1(ss,uu),ss=-0.001..0.001,uu=-0.001..0.001,color=gray):

> display(pf1,pp2,axes=boxed);

7.6.2 正規方程式による解法

デザイン行列へのデータ変換

> bb:=Vector(25): A:=Matrix(25,6):

p1:=2:

for i from 1 to 25 do

A[i,1]:=1;

A[i,2]:=x_i;

A[i,3]:=y_i;

A[i,4]:=x_i*y_i;

A[i,5]:=x_i^2;

A[i,6]:=y_i^2;

bb_i:=z_i;

end do:

正規方程式の解

> MatrixInverse(Transpose(A).A).(Transpose(A).bb);
−9.185257196× 10−13

−0.00000639644675999994798
0.00000639644220000032532

−5.45955358336000173
25.7696284050857187
25.7696284050857543



7.7 課題

1. 1次元の線形最小二乗法

次の 4点のデータを y = a1 + a2x+ a3x
2 で近似せよ (2006年度期末試験)．

X:=[0,1,2,3];

Y:=[1,3,4,10];

2. 2次元の最小二乗フィット

以下のデータを
f(x, y) = a1 + a2x+ a3y + a4xy

で近似せよ

x, y, z

-1, -1, 2.00000

-1, 0, 0.50000

-1, 1, -1.00000

0, -1, 0.50000

0, 0, 1.00000

0, 1, 1.50000

1, -1, -1.00000

1, 0, 1.50000

1, 1, 4.00000

第 8章

非線形最小 2乗法 (NonLinearFit)

8.1 非線形最小 2乗法の原理

前章では，データに近似的にフィットする最小二乗法を紹介した．ここでは，フィット式が多項式のような線形関

係にない関数の最小二乗法を紹介する．図のようなデータにフィットする場合を考えよう．

このデータにあてはめるのはローレンツ関数，

F (x;a) = a1 +
a2

a3 + (x− a4)
2

である．この関数の特徴は，今まで見てきた関数と違いパラメータが線形関係になっていない．誤差関数は，いまま

でと同様に

χ2 (a) =
N∑
i

d2i =
N∑
i

(F (xi; a)− yi)
2

で，a = a0, a1, ..をパラメータとして変えた時に最小となる値を求める点もかわらない．しかし，線形の最小二乗法

のように微分しても一元の方程式にならず，連立方程式を単に解くだけでは求まらない．

そこで図のような 2次関数の最小値を求める場合を考える．最小値の点 a0 のまわりで，Taylor展開すると，d,D

をそれぞれの係数とすると，

χ2 (a) = χ2 (a0)− d (a− a0) +
1

2
D (a− a0)

2

である．最小の点 a0 は，微分が 0になるので，

a0 = a+D−1 × (−d)

と予測される．図を参照して上の式を導け．またその意味を考察せよ．

現実には高次項の影響で計算通りにはいかず，単に最小値の近似値を求めるだけである．これは，χ (a)
2 の微分関

数の解を Newton 法で求める操作に対応する．つまり，この操作を何度も繰り返せばいずれ解がある精度で求まる

はず．

8.2 具体的な手順

パラメータの初期値を
a0 +∆ a, b0 +∆ b, c0 +∆ c, d0 +∆ d

とする．このとき関数 f を真値 a0, b0, c0, d0 のまわりでテイラー展開し，高次項を無視すると

∆ f = f (a0 +∆ a1, b0 +∆ b1, c0 +∆ c1, d0 +∆ d1)− f (a0, b0, c0, d0)

=

(
∂

∂a
f

)
0

∆a1 +

(
∂

∂b
f

)
0

∆b1 +

(
∂

∂c
f

)
0

∆c1 +

(
∂

∂d
f

)
0

∆d1

となる．

課題でつくったデータは t = 1から t = 256までの時刻に対応したデータ点 f1, f2, · · · f256 とする．各測定値とモ

デル関数から予想される値との差 ∆f1,∆f2, · · · ,∆f256 は，
∆f1
∆f2
...

∆f256

 = J


∆a1
∆b1
∆c1
∆d1

 (8.1)

となる．ここで J はヤコビ行列と呼ばれる行列で，4列 256行

J =


(

∂
∂af

)
1

(
∂
∂bf
)
1

(
∂
∂cf
)
1

(
∂
∂df

)
1

...
...

...
...(

∂
∂af

)
256

(
∂
∂bf
)
256

(
∂
∂cf
)
256

(
∂
∂df

)
256

 (8.2)

である．このような矩形行列の逆行列は転置行列 JT を用いて，‘

J−1 =
(
JTJ

)−1
JT (8.3)

と表わされる．したがって，真値からのずれは
∆a2
∆b2
∆c2
∆d2

 =
(
JTJ

)−1
JT


∆f1
∆f2
...

∆f256

 (8.4)

で求められる．理想的には (∆a2, ∆b2, ∆c2, ∆d2)は (∆a, ∆b, ∆c, ∆d)に一致するはずだが，測定誤差と高次項の

ために一致しない．初期値に比べ，より真値に近づくだけ．そこで，新たに得られたパラメータの組を新たな初期値

に用いて，より良いパラメータに近付けていくという操作を繰り返す．新たに得られたパラメータと前のパラメータ

との差がある誤差以下になったところで計算を打ち切り，フィッティングの終了となる．

8.3 Mapleによる解法の指針

線形代数計算のためにサブパッケージとして LinearAlgebraを呼びだしておく．

> restart;

with(plots):

with(LinearAlgebra):

データを読み込む．

> ndata:=8:

f1:=t->subs({a1=1,a2=10,a3=1,a4=4},a1+a2/(a3+(t-a4)^2) );

f1 := t 7→ 1 + 10
(
1 + (t− 4)

2
)−1

データの表示

> T:=[seq(f1(i),i=1..ndata)]:

listplot(T);

l1:=listplot(T):

ローレンツ型の関数を仮定し，関数として定義．

> f:=t->a1+a2/(a3+(t-a4)^2); nparam:=4:

f := t 7→ a1 +
a2

a3 + (t− a4 )
2

ヤコビアンの中の微分を新たな関数として定義．

> for i from 1 to nparam do

dfda||i:=unapply(diff(f(x),a||i),x);

end do;

dfda1 := x 7→ 1

dfda2 := x 7→
(
a3 + (x− a4 )

2
)−1

dfda3 := x 7→ − a2(
a3 + (x− a4 )

2
)2

dfda4 := x 7→ − a2 (−2x+ 2 a4 )(
a3 + (x− a4 )

2
)2

ここで，”||”は連結作用素とよばれるMapleのコマンドで，dfda||1 7→ dfda1と連結する．初期値を仮定して，デー

タとともに関数を表示．

> g1:=Vector([1,8,1,4.5]):

guess1:={}:

for i from 1 to nparam do

guess1:={op(guess1),a||i=g1[i]};

end do:

guess1;

{a1 = 1, a2 = 8, a3 = 1, a4 = 4.5}

> p1:=plot(subs(guess1,f(x)),x=1..ndata):

display(l1,p1);

見やすいように，小数点以下を 3桁表示に制限する．

> interface(displayprecision=3):

> df:=Vector([seq(subs(guess1,T[i]-f(i)),i=1..ndata)]);

df :=



0.396
0.897
2.538
3.600
−1.400
−0.462
−0.103
−0.016



> Jac:=Matrix(ndata,nparam):

for i from 1 to ndata do

for j from 1 to nparam do

Jac[i,j]:=evalf(subs(guess1,dfda||j(i)));

end do:

end do:

Jac; 

1.0 0.075 −0.046 −0.319
1.0 0.138 −0.152 −0.761
1.0 0.308 −0.757 −2.272
1.0 0.800 −5.120 −5.120
1.0 0.800 −5.120 5.120
1.0 0.308 −0.757 2.272
1.0 0.138 −0.152 0.761
1.0 0.075 −0.046 0.319



> tJac:=(MatrixInverse(Transpose(Jac).Jac)).Transpose(Jac);

tJac :=


0.565 0.249 −0.354 0.040 0.040 −0.354 0.249 0.565
−2.954 −0.506 4.012 −0.552 −0.552 4.012 −0.506 −2.954
−0.352 −0.029 0.557 −0.176 −0.176 0.557 −0.029 −0.352
−0.005 −0.012 −0.035 −0.080 0.080 0.035 0.012 0.005



> g2:=tJac.df;

g1:=g1+g2;

g2 :=


−0.235
5.592
0.613
−0.520



g1 :=


0.765
13.592
1.613
3.980


これをまたもとの近似値 (guess)に入れ直して表示させると以下のようになる．カーブがデータに近づいているのが

確認できよう．この操作をずれが十分小さくなるまで繰り返す．

> guess1:={seq(a||i=g1[i],i=1..nparam)};

p1:=plot(subs(guess1,f(x)),x=1..ndata):

display(l1,p1);

guess1 := {a1 = 0.765, a2 = 13.592, a3 = 1.613, a4 = 3.980}

4回ほど繰り返すと以下の通り，いい値に収束している．

guess1 := {a1 = 1.006, a2 = 9.926, a3 = .989, a4 = 4.000}

8.4 Gauss-Newton法に関するメモ

この Gauss-Newton法と呼ばれる非線形最小二乗法は線形問題から拡張した方法として論理的に簡明であり，広く

使われている．しかし，収束性は高くなく，むしろ発散しやすいので注意が必要．2次の項を無視するのでなく，う

まく見積もる方法を用いたのが Levenberg-Marquardt法である．明快な解説が Numerical Recipes in C(Ｃ 言語に

よる数値計算のレシピ）WilliamH.Press 他著，技術評論社 1993にある．

8.5 課題

1. 一山ピークへのフィット

以下の 256個のデータ

> ndata:=256; f1:=t->subs({a1=10,a2=40000,a3=380,a4=128},a1+a2/(a3+(t-a4)^2) );

> T:=[seq(f1(i)*(0.6+0.8*evalf(rand()/10^12)),i=1..ndata)]:

> f:=t->a1+a2/(a3+(t-a4)^2);

で近似したときのパラメータ a1,a2,a3,a4を求めよ．ただし，パラメータの初期値は，ある程度近い値にしな

いと収束しない．

2. 二山ピークのフィット以下のように作成したデータ

> ndata:=256; f1:=t->subs({a=10,b=40000,c=380,d=128},a+b/(c+(t-d)^2) );

> f2:=t->subs({a=10,b=40000,c=380,e=90},a+b/(c+(t-e)^2) );

> T:=[seq((f1(i)+f2(i))*(0.6+0.2*evalf(rand()/10^12)),i=1..ndata)]:

を

> f:=t->a1+a2/(a3+(t-a4)^2)+a2/(a3+(t-a5)^2);

f := t 7→ a1 +
a2

a3 + (t− a4 )
2 +

a2

a3 + (t− a5 )
2

で近似したときのパラメータを求めよ．

> l1:=listplot(T): display(l1);

8.6 解答例

2. ふた山ピークへのフィット．

> restart; with(plots): with(LinearAlgebra):

> f1:=t->subs({a=10,b=40000,c=380,d=128},a+b/(c+(t-d)^2) );

> f2:=t->subs({a=10,b=40000,c=380,e=90},a+b/(c+(t-e)^2) );

> T:=[seq((f1(i)+f2(i))*(0.6+0.2*evalf(rand()/10^12)),i=1..256)]:

f1 := t 7→ 10 + 40000
(
380 + (t− 128)

2
)−1

f2 := t 7→ 10 + 40000
(
380 + (t− 90)

2
)−1

> l1:=listplot(T):

> f:=t->a1+a2/(a3+(t-a4)^2)+a2/(a3+(t-a5)^2);

nparam:=5:

f := t 7→ a1 +
a2

a3 + (t− a4 )
2 +

a2

a3 + (t− a5 )
2

> for i from 1 to nparam do

dfda||i:=unapply(diff(f(x),a||i),x);

end do;

dfda1 := x 7→ 1

dfda2 := x 7→
(
a3 + (x− a4 )

2
)−1

+
(
a3 + (x− a5 )

2
)−1

dfda3 := x 7→ − a2(
a3 + (x− a4 )

2
)2 − a2(

a3 + (x− a5 )
2
)2

dfda4 := x 7→ − a2 (−2x+ 2 a4 )(
a3 + (x− a4 )

2
)2

dfda5 := x 7→ − a2 (−2x+ 2 a5 )(
a3 + (x− a5 )

2
)2

> g1:=Vector([10,1200,10,125,90]);

g1 :=


10
1200
10
125
90



> guess1:={seq(a||i=g1[i],i=1..nparam)};

guess1 := {a1 = 10, a2 = 1200, a3 = 10, a4 = 125, a5 = 90}

> p1:=plot(subs(guess1,f(x)),x=1..256):

display(l1);

> df:=Vector([seq(subs(guess1,T[i]-f(i)),i=1..256)]):

Jac:=Matrix(1..256,1..nparam,sparse):

for i from 1 to 256 do

for j from 1 to nparam do

Jac[i,j]:=evalf(subs(guess1,dfda||j(i)));

end do:

end do:

tJac:=(MatrixInverse(Transpose(Jac).Jac)).Transpose(Jac):

g2:=tJac.df; g1:=g1+g2;

g2 :=


−0.390553882992161205
1584.55290636967129
24.9577909601538366

−0.0472041829705451138
−0.00719532042503852940



g1 :=


13.6348019182603064
29567.3667677707381
410.545681677467769
128.512734548828887
90.9223109918718678



> guess1:={seq(a||i=g1[i],i=1..nparam)};

p1:=plot(subs(guess1,f(x)),x=1..256):

display(l1,p1);

guess1 := {a1 = 30.251, a2 = 3854.136, a3 = 39.571, a4 = 124.800, a5 = 89.960}

何回か繰り返せば，データ点に近づいてくるはず．

FFT(Fast Fourier Transformation)

9.1 FFTの応用

Fast Fourier Transformation(FFT)高速フーリエ変換 (あるいはデジタル (離散)フーリエ変換 (DFT))は，周波

数分解やフィルターを初め，画像処理などの多くの分野で使われている．基本となる考え方は，直交基底による関数

の内挿法である．最初にその応用例を見た後，どのような理屈で FFTが動いているかを解説する．

9.1.1 周波数分解

はじめの例は，周波数分解．先ずは，非整合な波を二つ用意しておく．

> restart:

funcs:=[sin(i/13),sin(i/2)];

#funcs:=[sin(i*2),2*sin(i/2)];

plot(funcs,i=0..300);

これを重ねあわせた波を作る．

> data1:=[]:

for i from 1 to 256 do

data1:=[op(data1),evalf(funcs[1]+funcs[2]))];

end do:

with(plots):

listplot(data1);

ゆっくり変化する波に，激しく変化する波が重なっていることが読み取れる．これに FFTを掛ける

> X:=array(data1):

Y:=array(1..256,sparse):

FFT(8,X,Y);

256

その強さを求めて，周波数で表示すると，

> Data2:=[seq([i,sqrt(X[i]^2+Y[i]^2)],i=1..128)]:

plot(Data2);

もとの 2つの周波数に対応するところにピークができているのが確認できる．広がりは，誤差のせい．logplotでも

良い．

9.1.2 高周波フィルター

次の例は，高周波フィルター．たとえば次のようなローレンツ関数を考える．

> restart;

f1:=t->subs(a=10,b=40000,c=380,d=128,a+b/(c+(t-d)^2));

f1 := t 7→ 10 +
40000

380 + (t− 128)
2

> plot(f1(t),t=1..256);

これにノイズがのると，次のようになる．

> T:=[seq(f1(i)*(0.6+0.8*evalf(rand()/10^12)),i=1..256)]:

#T:=[seq(evalf(rand()/10^12),i=1..256)]: #これはホワイトノイズ

#T:=[seq(f1(i),i=1..256)]: #これは元の関数そのまま

with(plots):

listplot(T);

これに高周波フィルターを掛けるとノイズが消えるが，その様子を示そう．先ずは，FFTを掛ける．

> Idata:=array([seq(0,i=1..256)]):

Rdata:=convert(T,array):

FFT(8,Rdata,Idata);

256

これは次のような強度分布をもっている．

> Adata:=[seq([i,sqrt(Idata[i]^2+Rdata[i]^2)],i=1..128)]:

> logplot(Adata);

低周波の部分に，ゆっくりとした変化を表す成分が固まっている．次のような三角フィルターを用意する．これは，

低周波ほど影響を大きくするフィルター．

> filter:=x->piecewise(x>=0 and x<=20,(1-x/20)): #三角フィルター

#filter:=x->piecewise(x>=0 and x<=20,1); #方形フィルター

plot(filter(x),x=0..40);

これとデータを各点で掛けあわせる事によって，フィルターを通したことになる．

> FRdata:=array([seq(Rdata[i]*filter(i),i=1..256)]):

> FIdata:=array([seq(Idata[i]*filter(i),i=1..256)]):

先ほどと同様に表示すると

> Bdata:=[seq([i,sqrt(FIdata[i]^2+FRdata[i]^2)],i=1..128)]:

> logplot(Adata);

i = 20以上の領域がフィルターによってちょん切られていることが確認できる．これを逆フーリエ変換する．

> iFFT(8,FRdata,FIdata);

256

これを表示すると，

> listplot(FRdata);

となる．ノイズが取り除かれているのが確認できる．元の関数に加えたホワイトノイズに FFTを掛ければ分かるが，

全周波数域にわたって均質に広がった関数となる．これを三角フィルターなどで高周波成分をカットすることで，ノ

イズが取り除かれていくのが理解されよう．

9.2 FFTの動作原理

このように便利な FFTであるが，どのような理屈で導かれるのか？　 Fourier変換法は，この課題だけでも何回

ものコマ数が必要なほどの内容を含んでいる．ここでは，その基本となる考え方（のひとつ)だけを提示する．

1. 関数の内挿で導入した基底関数を直交関数系でとる．ところが，展開係数を逆行列で求める手法では計算が

破綻．

2. 直交関係からの積分による係数決定．

3. 選点直交性による計算の簡素化．

4. 高速フーリエ変換アルゴリズムによる高速化．

9.3 関数内挿としての Fourier関数系

一連の関数系による関数の内挿は，基底関数を φn(x)として

F (x) =
N∑

n=1

anφn(x)

で得られることを見た．Fourier 変換では基底関数として φn (x) = sin (2πnx ) , cos (2πnx ) をとる．関数の内挿法

で示したように，この xi での値 fi, i = 1 · · ·M と，近似の次数 (N)とでつくる係数行列，

A =

 φ0(x0) φ1(x0) · · · φN (x0)
...

...
...

...
φ0(xM ) φ1(xM ) · · · φN (xM )


を求めて，係数 ai とデータ点 fi をそれぞれベクトルと考えると，

A.a = f

から，通常の逆行列を求める手法で係数を決定することもできる．しかし，この強引な方法はデータ数，関数の次数

が多い，フーリエ変換が対象としようとする問題では破綻する．もっといい方法が必要で，それが直交関数系では存

在する．

9.4 直交関係からの積分による係数決定

関数の直交関係は， ∫ b

a

φn (x)φm (x) dx = δmnCn =

{
Cm at n = m
0 at n ̸= m

(9.1)

である．定数 Cm は，sin, cosの三角関数系では次の通り．

> plot([sin(x),sin(3*x)],x=0..2*Pi);

> plot([sin(x)*sin(3*x)],x=0..2*Pi, color=black);

> int(sin(x)*sin(3*x),x=0..2*Pi);

0

> for i from 1 to 3 do for j from 1 to 3 do S:=int(sin(i*x)*sin(j*x),x=0..2*Pi);

> print(i,j,S); end do; end do:

1, 1, π

1, 2, 0

1, 3, 0

2, 1, 0

2, 2, π

2, 3, 0

3, 1, 0

3, 2, 0

3, 3, π

∫ b

a

F (x)φm (x) dx

を考える．先程の 9.1式をいれると∫ b

a

F (x)φm (x) dx =

∫ b

a

N∑
n=1

anφn (x)φm (x) dx =

{
amCm at n = m
0 at n ̸= m

(9.2)

となる．こうして，係数 an が

an =
1

Cn

∫ b

a

F (x)φn (x) dx

で決定できる．

9.5 直接積分によるフーリエ係数

対象とする関数をまず作る．

> restart;

> #F:=x->piecewise(x=0,1/2,x>0,x);

#F:=x->piecewise(x<1/2,x,x>=1/2,1-x);

#F:=x->piecewise(x<1/2,-1,x>1/2,1);

F:=x->piecewise(x<1/2,-1,x>=1/2,1);

#F:=x->piecewise(x>0 and x<1/2,-1,x>1/2,1);

#F:=x->x-1/2;

plot(F(x),x=0..1);

F := x 7→ piecewise(x <
1

2
,−1,

1

2
≤ x, 1)

piecewise関数は階段関数で，振る舞いはコメント (#)を適当に外して確認せよ．初期設定．

> KK:=3; N:=2^KK;L:=1-0;

> 2*Pi*1/L*x;

2πx

> int(F(x)*cos(2*Pi*1/L*x),x=0..L);

0

> for n from 0 to N do

a[n]:=2/L*int(F(x)*cos(2*Pi*n/L*x),x=0..L);

end do;

a0 := 0

a1 := 0

a2 := 0

a3 := 0

a4 := 0

a5 := 0

a6 := 0

a7 := 0

a8 := 0

> for n from 0 to N do

b[n]:=2/L*int(F(x)*sin(2*Pi*n/L*x),x=0..L);

end do;

b0 := 0

b1 :=
4

π
b2 := 0

b3 :=
4

3π
b4 := 0

b5 :=
4

5π
b6 := 0

b7 :=
4

7π
b8 := 0

ここで，オイラーの関係

a[n] = c[n] + c[−n], b[n] = I(c[n]− c[−n])

c[−n] =
1

2
(a[n] + b[n]), c[n] =

1

2
(a[n]− Ib[n]) (9.3)

を使って，三角関数系から expへ変換する．

> for n from 0 to N do c[n]:=1/L*int(F(x)*exp(-I*2*Pi*n/L*x),x=0..L); end do;

> for n from 1 to N do c[-n]:=1/L*int(F(x)*exp(I*2*Pi*n/L*x),x=0..L); end do;



92 第 9章 FFT(Fast Fourier Transformation)

c0 := 0

c1 :=
2I

π
c2 := 0

c3 :=
2I

3π
c4 := 0

c5 :=
2I

5π
c6 := 0

c7 :=
2I

7π
c8 := 0

c−1 := −2I

π
c−2 := 0

c−3 := − 2I

3π
c−4 := 0

c−5 := − 2I

5π
c−6 := 0

c−7 := − 2I

7π
c−8 := 0

> F1:=unapply(sum(evalf(c[i]*exp(I*2*Pi*i/L*x)),i=-(N-1)..(N-1)),x):

> plot({Re(F1(x)),F(x)},x=0..1);

> evalf(2/3/Pi);

0.2122065907

9.6 選点直交性による計算の簡素化

ところが，実際に積分していては，時間がかかりすぎる．直交関数系の選点直交性を使うとより簡単になる．

■直交関数系の選点直交性

直交多項式は，
φn (x) = 0 at x1, x2, · · ·xn

である．n− 1以下の次数m, lでは，

n∑
i=1

ϕl (xi)φm (xi) = δmlCl

が成り立つ．これは，直交関係と違い積分でないことに注意．証明は略．

これを使えば，この先程の直交関数展開

F (x) =
N∑
l=1

alφl (x)

の両辺に φm (xi)を掛けて iについて和をとれば，

n∑
i=1

F (xi)ϕm (xi) = (9.4)

n∑
i=1

N∑
l=1

alφl (xi)φm (xi)

=
N∑
l=1

al

n∑
i=1

φl (xi)φm (xi)

=
N∑
l=1

alδmlCm = amCm

となる．つまり，

am =
1

Cm

n∑
i=1

F (xi)φm (xi)

となり，単純な関数の代入とかけ算で係数が決定される．

9.6.1 選点直交性を用いた結果

> KK:=4; N:=2^KK;L:=1-0;

> for k from 0 to N-1 do

c[k]:=evalf(sum(F(i*L/N)*exp(-I*2*Pi*k*i/N),i=0..N-1));

end do;

c_0:=0.

c_1:=-2.000000000 + 10.05467898 I

c_2:=0.

c_3:=-2.000000000 + 2.993211524 I

c_4:=0.

c_5:=-2.000000001 + 1.336357276 I

c_6:=0.

c_7:=-2.000000001 + 0.3978247331 I

c_8:=0.

c_9:=-2.000000001 - 0.3978247331 I

c_10:=0.

c_11:=-2.000000001 - 1.336357276 I

c_12:=0.

c_13:=-2.000000000 - 2.993211524 I

c_14:=0.

c_15:=-2.000000000 - 10.05467898 I

> F1:=unapply(sum(evalf(c[i]*exp(I*2*Pi*i/L*x)/N),i=0..(N/2-1))+

> sum(evalf(c[N-i]*exp(-I*2*Pi*i/L*x)/N),i=1..(N/2-1)),x):

> plot({Re(F1(x)),F(x)},x=0..1);

9.7 高速フーリエ変換アルゴリズムによる高速化

sin, cosと exp関数を結びつけるオイラーの関係を使うと，

exp

(
2π

N
I

)
= cos

(
2π

N

)
+ I sin

(
2π

N

)
と変換できる．これを使うと，

ck =
1

Cm

N−1∑
i=0

F (xi) exp

(
−2πI

N

)

となる．N = 8の場合を実際に計算すると，z = exp(−2π

8
I)として，z8 = 1, z9 = z, · · · を使うと，



c0
c1
c2
c3
c4
c5
c6
c7


=



1 1 1 1 1 1 1 1
1 z z2 z3 z4 z5 z6 z7

1 z2 z4 z6 1 z2 z4 z6

1 z3 z6 z z4 z7 z2 z5

1 z4 1 z4 1 z4 1 z4

1 z5 z2 z7 z4 z1 z6 z3

...

...





F0

F1

F2

F3

F4

F5

F6

F7


となる．この行列計算を素直に実行すると，8 × 8=64 回の演算が必要となる．これを減らせないかと考えたのが，

高速フーリエ変換の始まりである．この行列をよく見ると同じ計算を重複しておこなっていることが分かる．そこ

で，行列の左側と右側で同じ計算をしている部分をまとめると，

c0
c1
c2
c3
c4
c5
c6
c7


=



1 1 1 1 0 0 0 0
0 0 0 0 1 z z2 z3

1 z2 z4 z6 0 0 0 0
0 0 0 0 1 z3 z6 z
1 z4 1 z4 0 0 0 0
0 0 0 0 1 z5 z2 z7

1 z6 z4 z2 0 0 0 0
0 0 0 0 1 z7 z6 z5





F0 + F4

F1 + F5

F2 + F6

F3 + F7

F0 − F4

F1 − F5

F[ ] − F6

F[ ] − F7


とすることができる．ここで，z4 = −1などを使っている．右側のベクトルの計算でロスするが，行列の中の計算の

回数を半分に減らすことができる．再度できあがった行列を見れば，同じ計算をさらにまとめることができそうであ

る．こうして，次々と計算回数を減らしていくことが可能で，最終的に行列部分の計算が一切なくなる．残るのは，

右側のベクトルの足し算引き算だけになる．

このベクトルの組み合わせは，一見相当複雑そうで，その条件分岐で時間がかかりそうに思われる．しかし，よく

調べてみれば，単純なビット演算で処理することが可能であることが判明した．こうして，2の整数乗のデータの組

に対しては，極めて高速にフーリエ変換を実行することが可能となった．FFTでの演算回数は，データ数を Nとす

ると
N log2 N

となる．単純な場合の N2 と比較すると，以下のようになり，どれだけ高速化されているかが理解されよう．

> dN2:=[]; dFft:=[]; for i from 2 to 16 do N:=2^i; n2:=N*N; Fft:=N/2*log[2](N);

> Fft/n2; printf("%10d %12d %12d %10.5f\n",N,n2,Fft,evalf(Fft/n2));

> dN2:=[op(dN2),[N,n2]]; dFft:=[op(dFft),[N,Fft]]; end do:

4 16 4 0.25000

8 64 12 0.18750

16 256 32 0.12500

32 1024 80 0.07812

64 4096 192 0.04688

128 16384 448 0.02734

256 65536 1024 0.01562

512 262144 2304 0.00879

1024 1048576 5120 0.00488

2048 4194304 11264 0.00269

4096 16777216 24576 0.00146

8192 67108864 53248 0.00079

16384 268435456 114688 0.00043

32768 1073741824 245760 0.00023

65536 4294967296 524288 0.00012

> with(plots):

> l1:=plot(dN2): l2:=plot(dFft):

> display(l1,l2);

> l1:=logplot(dN2): l2:=logplot(dFft):

> display(l1,l2);



96 第 9章 FFT(Fast Fourier Transformation)

> l1:=loglogplot(dN2): l2:=loglogplot(dFft):

> display(l1,l2);

9.8 FFT関数を用いた結果

> KK:=4;

N:=2^KK;i:=’i’;L:=1-0;

x1:=array([evalf(seq(F(i/N),i=0..N-1))]);

y1:=array([evalf(seq(0,i=0..N-1))]);

x1 :=
[
−1.0 −1.0 −1.0 −1.0 −1.0 −1.0 −1.0 −1.0 1.0 1.0 1.0 1.0 1.0 1.0 1.0 1.0

]
y1 :=

[
0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0 0.0

]

> FFT(KK,x1,y1);

16

> interface(displayprecision=2):

print(x1);print(y1);

[
0.0 −2.0 0.0 −2.0 0.0 −2.0 0.0 −2.0 0.0 −2.0 0.0 −2.0 0.0 −2.0 0.0 −2.0

][
0.0 10.05 0.0 2.99 0.0 1.34 0.0 0.40 0.0 −0.40 0.0 −1.34 0.0 −2.99 0.0 −10.05

]

> F2:=unapply(sum(evalf((x1[i]+I*y1[i])*exp(I*2*Pi*(i-1)/L*x)/N),i=1..N/2)+

sum(evalf((x1[N-i+2]+I*y1[N-i+2])*exp(-I*2*Pi*(i-1)/L*x)/N),i=2..N/2),x):

> plot({Re(F2(x)),F(x)},x=0..1);

9.9 課題と解答例

課題 下の例に従って，sin(i/13)と sin(i/2)の合成波を作成し，FFTをかけた後，周波数での強度を表示せよ．合

成波 (2*sin(i*2)+sin(i/2))との違いをのべよ．

解答例 restart:

funcs:=[sin(i/13),sin(i/2)];

#funcs:=[2*sin(i*2),sin(i/2)];

plot(funcs,i=0..300);

funcs := [sin (i/13) , sin (i/2)]

> data1:=[]:

for i from 1 to 256 do

data1:=[op(data1),

evalf(funcs[1]+funcs[2])];

# evalf(funcs[1])];

end do:

with(plots):

listplot(data1);

> X:=array(data1):

Y:=array(1..256,sparse):

FFT(8,X,Y);

256

> Data2:=[seq([i,sqrt(X[i]^2+Y[i]^2)],i=1..128)]:

plot(Data2);

